

# Programming in GILDAS

Major revision on October 2008, but still not fully up-to-date

Questions? Comments? Bug reports? Mail to: `gildas@iram.fr`

The GILDAS team welcomes an acknowledgment in publications  
using GILDAS software to reduce and/or analyze data.

Please use the following reference in your publications:

<http://www.iram.fr/IRAMFR/GILDAS>

## Documentation

In charge: S. Guilloteau<sup>1</sup>.

Active developers: S. Bardeau<sup>2</sup>, J. Pety<sup>2,3</sup>, E. Reynier<sup>2</sup>.

## Software (GILDAS kernel)

In charge: J. Pety<sup>2,3</sup>.

Active developers: S. Bardeau<sup>2</sup>, S. Guilloteau<sup>1</sup>, E. Reynier<sup>2</sup>.

Main past contributors: F. Badia, D. Broguière, G. Buisson, L. Desbats, G. Duvert,  
T. Forveille, R. Gras, P. Valiron.

1. Observatoire de Bordeaux
2. IRAM
3. Observatoire de Paris

Related information are available in

- GILDAS general introduction
- GREG: Graphical Possibilities
- SIC: Command Line Interpreter



# Contents

<b>Introduction</b>	<b>vii</b>
<b>1 SIC Programming Manual - Partially updated on Oct.2008</b>	<b>1</b>
1.1 Introduction . . . . .	1
1.2 Initializing SIC: Languages and Packages . . . . .	2
1.2.1 Packages . . . . .	2
1.2.2 Language Definition . . . . .	4
1.2.3 The command dispatching and handling . . . . .	6
1.3 The Help File . . . . .	7
1.4 Retrieving Arguments . . . . .	7
1.5 The messaging facility . . . . .	8
1.5.1 Writing on files: Fortran logical unit number . . . . .	9
1.5.2 Linking on Linux: TO BE UPDATED . . . . .	9
1.6 The Library Version . . . . .	9
1.6.1 Library Only mode . . . . .	10
1.7 Using Variables . . . . .	10
1.7.1 Definition . . . . .	10
1.7.2 Assignment and Examination . . . . .	11
1.7.3 Mathematical Formula Handling . . . . .	13
1.7.4 Deleting Variables . . . . .	13
1.8 Using Functions . . . . .	13
1.9 SIC Callable Routines . . . . .	13
1.9.1 Monitor interface routines . . . . .	13
1.9.2 SIC Arguments Retrieving Routines . . . . .	15
1.9.3 Command Line Interpreter Subroutines . . . . .	17
1.9.4 All Purpose General Subroutines . . . . .	18
1.9.5 Symbol Manipulation Routines . . . . .	19
1.10 Obsolescent Features . . . . .	19
1.10.1 Function . . . . .	19
1.10.2 Routines for Library usage (Obsolescent, for record only) . . . . .	19
<b>2 GreG Programming Manual</b>	<b>21</b>
2.1 Interaction with FORTRAN programs . . . . .	22
2.2 Basic Routines . . . . .	22
2.3 Linking . . . . .	23
2.3.1 UNIX systems . . . . .	23
2.4 Running . . . . .	24

2.5	Example . . . . .	24
2.6	Array Transfer . . . . .	24
2.6.1	GR4_GIVE - GR8_GIVE . . . . .	24
2.6.2	GR4_GET - GR8_GET . . . . .	24
2.6.3	GR4_RGIVE - GR8_RGIVE . . . . .	25
2.6.4	GR4_LEVELS - GR8_LEVELS . . . . .	25
2.7	Immediate Routines . . . . .	25
2.7.1	GR_SEGM . . . . .	26
2.7.2	GR_OUT . . . . .	26
2.7.3	DRAW - RELOCATE . . . . .	26
2.7.4	GDRAW - GRELOCATE . . . . .	26
2.7.5	GR4_PHYS_USER - GR8_PHYS_USER . . . . .	26
2.7.6	GR4_USER_PHYS - GR8_USER_PHYS . . . . .	26
2.7.7	GR4_CONNECT - GR8_CONNECT . . . . .	26
2.7.8	GR4_HISTO - GR8_HISTO . . . . .	27
2.7.9	GR4_MARKER - GR8_MARKER . . . . .	27
2.7.10	GR4_CURVE - GR8_CURVE . . . . .	27
2.7.11	GR4_EXTREMA - GR8_EXTREMA . . . . .	28
2.7.12	GR8_BLANKING . . . . .	28
2.7.13	GR8_SYSTEM - GR8_PROJEC . . . . .	28
2.7.14	GR4_RVAL . . . . .	28
2.7.15	GR_WHERE . . . . .	29
2.7.16	GR8_TRI . . . . .	29
2.7.17	GR8_SORT . . . . .	29
2.7.18	GR_CLIP . . . . .	29
2.8	The cursor routine . . . . .	29
2.9	GREG High-Level Subroutines . . . . .	30
<b>3</b>	<b>Task Programming Manual</b>	<b>35</b>
3.1	General Outline and Data Structure of Images . . . . .	35
3.2	Fortran-90 access to images . . . . .	36
3.3	Obsolescent Fortran-77 access routines . . . . .	44
3.3.1	Image Slot Handling . . . . .	44
3.3.2	Image Connection . . . . .	45
3.3.3	Memory Connection . . . . .	46
3.3.4	Image Header Access . . . . .	47
3.4	Creating GILDAS Tasks . . . . .	51
3.5	A Template Task . . . . .	52
3.5.1	Source code . . . . .	52
3.5.2	Initialization file . . . . .	56
3.5.3	The HELP file . . . . .	56
3.5.4	Checker File . . . . .	57
3.6	Debugging Tasks . . . . .	58

<b>4</b>	<b>GTV Programming Manual</b>	<b>59</b>
4.1	Concept . . . . .	59
4.2	Programming . . . . .	60
4.3	Basic Sequence . . . . .	61
4.4	Plot Structuration and multi-window applications . . . . .	62
4.5	Subroutines . . . . .	62



# Introduction

The GILDAS Programming Manual includes all necessary information to create applications based on the GILDAS tools. The tools includes:

- SIC the command line interpreter, written in FORTRAN and callable as a subroutine by any program.
- GREG the high level Graphic library.
- Independent *Tasks* for specific applications. The *Task* programming section describes how to create new GILDAS tasks.

The GTVIRT low level Graphic library is also described for completeness, although very few programs will actually require using its possibilities directly.

This manual contains several chapters. Chapter 2 (the SIC programming manual) is essential when constructing interactive applications. Chapter 3 (the GREG programming manual) is essential for all graphics applications. To create tasks, the user only needs to read Chapter 4 (the GILDAS task programming manual). Sophisticated graphic application may require to read Chapter 5, the GTVIRT programming manual.





# Chapter 1

## SIC Programming Manual - Partially updated on Oct.2008

### 1.1 Introduction

SIC (\*) is a command line interpreter, written in FORTRAN and callable as a subroutine by any program. It provides a command language, with the following major features:

- resolution of command abbreviations
- definition of symbols
- macro capabilities with arguments substitution during execution
- log file
- multi-language structure
- loop buffers for repetitive actions
- variables, arithmetic and logical expressions evaluation
- structured logical tests
- error recovery
- stack buffer
- editing of command lines
- GUI interface on Motif, OS/X and MS-Windows systems
- optionally, a Python interface

This section concerns the programmer who wants to use SIC as the monitor of a simple or complex, evolutive, documented interactive program. It assumes that the reader is already familiar with all capabilities of SIC, so that he should be able to design a program (or better a system) around the SIC monitor.

It is in fact essentially a “CookBook” giving a list of recipes to build a program using SIC, or interface SIC with a preexisting ensemble of routines. Following these recipes should result in a complete success, that is, a program fulfilling the following requirements:

- Works in interactive or batch, reading commands on Standard Input This is called **Read Mode**.
- Can be called as a subroutine to execute a single command (**Execute Mode**). In case of error or if a **PAUSE** is encountered, the system switches to Read Mode for interactive error recovery or to read commands after the **PAUSE**.
- Is completed by a library version
- The library version can be used to make another language around it.
- Is portable on all Unix-like systems (including Mac OS/X) and MS-Windows.

The section 1.2 describes the command vocabulary structure, and the initialization sequence of SIC. Section 1.3 indicates how to write the corresponding **HELP** files. Section 1.4 describes how to retrieve arguments from a SIC command line. Section 1.6 indicates how to complete the program so that it can also be a library of routines. Finally, Section 1.7 indicates how to use SIC variables, and Section 1.9.1 give all entry points of the SIC monitor.

SIC programming interface is designed for the Fortran language. Interfacing with C language requires some system dependent precautions concerning argument passing mechanism, in particular for character strings.

## 1.2 Initializing SIC: Languages and Packages

SIC is a multi-package multi-language interpreter. A package is a collection of tools logically grouped together to provide some related applications, e.g. a plotting ensemble like GreG. A package can reference several languages: for example, GreG contains 4 languages: GREG1, GREG2, GREG3 for 1-D, 2-D and 3-D data plotting, and GTV for the basic plot actions. SIC itself has 5 languages: SIC, VECTOR, ADJUST, GUI and TASK, with the last two in “Library only” mode.

### 1.2.1 Packages

A main program using SIC is reduced to practically nothing: For example, the main program of Mapping is

```
program mapping
!-----
! Main astro program
!-----
external :: mapping_pack_set
!
call gmaster_run(mapping_pack_set)
!
end program mapping
```

All the job is done by `gmaster_run` which declares `Mapping` as the master program, and calls its package setup routine `mapping_pack_set`.

This package setup routine named `name_pack_set`, where `name` is the package name, contains e.g. for `MAPPING`

```

subroutine mapping_pack_set(pack)
  use gpack_def
  !
  type(gpack_info_t), intent(out) :: pack
  !
  external :: greg_pack_set
  external :: mapping_pack_init
  external :: mapping_pack_on_exit
  !
  pack%name='mapping'
  pack%ext = '.map'
  pack%depend(1:1) = (/ locwrdd(greg_pack_set) /)
  pack%init=locwrdd(mapping_pack_init)
  pack%on_exit=locwrdd(mapping_pack_on_exit)
  pack%authors="J.Pety, N.Rodriguez-Fernandez, S.Guilloteau, F.Gueth"
  !
end subroutine mapping_pack_set

```

pack is here a Fortran derived-type variable, with the following elements

- `pack%name` The package name (Character(len=12))
- `pack%ext` The default extension for procedures (Character(len=12))
- `pack%depend` The addresses of the package setting routines required by this package (an integer array of type Address.Length, 32 or 64 bits depending on the machine). Note that a package may depend on several other ones. The dependent packages are automatically loaded.
- `pack%init` The address of the initialization routine (an integer of type Address.Length)
- `pack%on_exit` The address of the cleaning on exit routine (an integer of type Address.Length)
- `pack%authors` A character string handling the authors name

The initialisation routine referred to by `pack%init` must setup all the package languages, and execute any code require to provide the initial setup of the package. In addition, it is recommended to setup the messaging facility: in the example below this is done using `name_message_set_id`.

```

subroutine mapping_pack_init(gpack_id,error)
  use sic_def
  !-----
  !
  !-----
  integer :: gpack_id
  logical :: error
  !
  ! Local language
  call init_clean
  !
  ! One time initialization
  call map_message_set_id(gpack_id) ! Set library id

```

```

!
! Language priorities
call exec_program('SIC'//backslash//'SIC PRIORITY 1 CLEAN')
!
! Specific initializations
!
end subroutine mapping_pack_init

```

### 1.2.2 Language Definition

A language contains an ensemble of commands. Each language must be initialized by means of an appropriate call to the routine SIC\_BEGIN. This routine has the following calling list:

SIC\_BEGIN(LANG,HELP,MCOM,VOCAB,VERSION,Dispatch,Error)

LANG is a Character constant which gives the name of the language. LANG need not be distinct from the command names, and it must not include the character \ which will appear in the internal HELP of SIC. MCOM is the number of commands and options appearing in the language vocabulary VOCAB. All commands in a SIC language are CHARACTER constants in which the first character is a reserved code. Lower case characters are **not** allowed, but the special character "-" may appear. All other special characters have (or may have in the future) some specific meaning in SIC. The following is the DATA initialization statement of SIC itself given as an example of language.

```

integer :: sic_commands
parameter (sic_commands=76)
character(len=12) :: sic_vocab(sic_commands)
data sic_vocab/
&
' HELP', & ! Must be the first one...
' @', & ! Others by alphabetic order
' ACCEPT', ' /ARRAY', ' /BINARY', ' /COLUMN', ' /FORMAT', ' /LINE', &
' BEGIN', &
' BREAK', &
' COMPUTE', ' /BLANKING', &
' CONTINUE', &
' DEFINE', ' /GLOBAL', ' /LIKE', &
' DELETE', ' /SYMBOL', ' /VARIABLE', ' /FUNCTION', &
' EDIT', &
' ELSE', &
' EXAMINE', ' /GLOBAL', ' /FUNCTION', ' /HEADER', ' /ADRESS', ' /ALIAS', &
'#EXIT', &
' END', &
' FOR', ' /WHILE', &
' IF', &
' IMPORT', ' /DEBUG', &
' LET', ' /NEW', ' /PROMPT', ' /WHERE', ' /RANGE', ' /CHOICE', ' /FILE', &
' /INDEX', ' /SEXAGESIMAL', ' /LOWER', ' /UPPER', ' /FORMAT', &
' /FORMULA', ' /REPLACE', ' /STATUS', &

```

```

' MESSAGE',                                &
' MFIT',      '/START', '/STEP', '/WEIGHTS', '/EPSILON', '/METHOD',      &
              '/ITERATIONS', '/QUIET', '/BOUNDS',                        &
' NEXT',                                &
' ON',                                &
' PAUSE',                                &
' PYTHON',    '/END',                  &
'#QUIT',                                &
'#RECALL',                                &
' RETURN',                                &
' SAY',      '/FORMAT',                &
' SIC',                                &
' SORT',                                &
' SYMBOL',    '/INQUIRE',              &
' SYSTEM',                                &
' TYPE' /

```

The first character code is interpreted as follow

<space> Usual command

/ Option of the preceding command

# Command forbidden in the stack and in procedures. Commands of this type are still written in the Log\_File.

\$ Special command which must not be inserted automatically in the stack. Commands of this type are written in the Log\_File

\* Purely informative command, which is only useful in an interactive session. Commands of this type are not written in the Log\_File and not inserted in the Stack.

The \* code can also be used for complex commands using alphanumeric keywords that you wish to expand yourself before saving them in the Log\_File and the Stack, or for commands that may require to output several records (See SIC\_INSERT, SIC\_LOG and SIC\_LIRE). The GREG command DRAW is an example of this.

The programmer should be careful about deciding what attribute to give to any command. All the options referring to a command must immediately follow it. At initialization time, SIC recognizes the options and set up a table of pointers connecting the options to their respective commands. A failure to respect the adequate order results in a very strange vocabulary.

VERSION is a character string which must contain the version number, and may contain the date of last modification, the name of the programmer,... The following example is recommended for optimum presentation (respect the alignment of the date for different languages):

```
VERSION='1.0    21-APR-1984    Programmer Name'
```

Dispatch is a subroutine to handle the dispatching of actions for all commands in the language.

ErrorRoutine is a subroutine to perform the error recovery action in case the execution of a command of this language returns an error. It may do nothing.

### 1.2.3 The command dispatching and handling

The dispatching routine is typically a big SELECT CASE based on the command name, like this one

```

subroutine run_clean (line,comm,error)
  use gbl_message
  !-----
  ! CLEAN Main routine
  !   Call appropriate subroutine according to COMM
  !-----
  character(len=*) , intent(inout) :: line ! Command line
  character(len=*) , intent(in)    :: comm ! Command name
  logical,          intent(out)    :: error ! Logical error flag
  !
  call map_message(seve%c,'CLEAN',line)
  !
  ! Analyze command
  select case (comm)
  case ('LOAD')
    call load_buffer(line,error)
  case ('READ')
    call read_image(line,error)
  case ('CLARK')
    call clark_clean(line,error)
  case ('FIT')
    ... etc...
  case default
    call map_message(seve%i,'CLEAN',comm//' not yet implemented')
  end select
  !
end subroutine run_clean

```

A handling routine for the a command looks like

```

SUBROUTINE COM1(LINE,ERROR)
  CHARACTER*(*) LINE
  LOGICAL ERROR
  LOGICAL SIC_PRESENT
  INTEGER IARG1_OPT1
  REAL ARG1
  !
  ! Test presence of option 1, and if so
  !   Decode Argument 1 of this option with a default value
  IF (SIC_PRESENT(1,0)) THEN
    IARG1_OPT1 = 10
    CALL SIC_I4 (LINE,1,1,IARG1_OPT1,.FALSE.,ERROR)
    IF (ERROR) RETURN
    WRITE (6,*) 'Option 1 Set With Argument',IARG1_OPT1
  !

```

```

        ENDIF
!
! Retrieves and decode first argument of the command
        CALL SIC_R4 (LINE,0,1,ARG1,.TRUE.,ERROR)
        IF (ERROR) RETURN
        WRITE (6,*) 'Command COM1 activated. ARG1',ARG1
! End of interface analysis, call a standard FORTRAN routine with
! all parameters now defined
        CALL SUB1(ARG1,ARG2,...,IARG1,...,ERROR)
        RETURN
        END

```

The command line buffer `LINE` must be passed by argument, never in a Module, to allow modularity and multi-language use.

## 1.3 The Help File

The `HELP` files are simple text files with two levels of help. The list of help topics in the library should be identical to the list of commands of the program. In addition, a specific topic named `LANGUAGE` should include a one line description of all commands, with a subtopic named `NEWS` which describe the latest news on the specific language.

The `HELP` files format is the following

```

1 TOPIC
    help for this topic
2 SUBTOPIC
    help for a subtopic of the previous topic
1 OTHER_TOPIC
    text for other_topic
1 ENDOFHELP

```

where 1 and 2 are in the first column of the text file, and followed by a single space. This format is easy to modify as it is a simple text file, but much slower to access because it is sequential. The `1 ENDOFHELP` (with no trailing characters) sequence indicates the end of the help file.

The help file name should be assigned to the logical name specified in the `SIC_BEGIN` call, or given explicitly.

## 1.4 Retrieving Arguments

To properly interface your program, you must know how to retrieve an argument from the command buffer processed by `SIC`. In standard `SIC` offers for each command up to 98 options, and for each option (or command) up to 98 arguments, the total number of arguments and options in a command line being limited to 99.

Arguments in `SIC` command lines are positional. The position of argument number `IARG` of option number `IOPT` of the current command is kept in `SIC` by means of internal pointers. By convention, argument number 0 refers to the option (or command) itself, and option number 0 to the command itself. Four standard routines are provided to decode the following type of arguments. All these routines do not modify the value of the argument if it is missing in the

command buffer. They may return an error condition and output an error message if requested in this case.

Argument Type	Routine Name	
Integer*4	SIC_I4	
Real*4	SIC_R4	
Real*8	SIC_R8	
Logical*4	SIC_L4	
Character	SIC_CH	
Character	SIC_KE	(Ucase converted keyword)

The first four routines have the following calling list

```
CALL SIC_name (LINE,IOPT,IARG,Arg,PRESENT,ERROR)
```

in which

LINE	The command buffer
IOPT	The option number (0=command)
IARG	The argument number (0=command or option)
Arg	The argument to be retrieved
PRESENT	A flag indicating whether an error must occur if the argument is not present in the command line
ERROR	An error flag set in case of decoding errors, or for missing argument when PRESENT is set.

SIC\_CH and SIC\_KE are slightly different, because they also return the length of the character string:

```
CALL SIC_CH (LINE,IOPT,IARG,Arg,Larg,PRESENT,ERROR)
CALL SIC_KE (LINE,IOPT,IARG,Arg,Larg,PRESENT,ERROR)
```

where

Larg	Integer, true length of Arg
------	-----------------------------

While SIC\_CH returns any character string, with implicit formatting if necessary, but no case conversion, SIC\_KE returns upcase keywords.

Two additional routines provide a way of testing the presence of an argument SIC\_PRESENT(IOPT,IARG) returns the logical value .TRUE. if the required argument is present, and SIC\_LEN(IOPT,IARG) returns the actual length of the argument (0 means the argument is missing). A last routine, SIC\_NARG(IOPT) indicates how many arguments are actually present for option IOPT.

Note that due to the structure of SIC vocabulary, IOPT is always the sequential number of the option as defined by the ordering in the DATA statement. Hence IOPT is not dependent on the order in which the options appear in the command buffer currently analysed.

SIC also includes some general character string manipulation routines which may be of interest in many other problems. All these routines are described more completely in section 1.9.1.

## 1.5 The messaging facility

This needs to be documented.



### 1.5.1 Writing on files: Fortran logical unit number

SIC has the following usage of Fortran logical unit numbers:

```

5          Input of Commands, shareable
6          Output of messages, shareable
50 to 99  Used by SIC_GETLUN

```

Units 5 and 6 are respectively accessed via usual `READ` and `WRITE` instructions.

Units 5 and 6 are respectively accessed via usual `READ` and `WRITE` instructions. If you wish to be able to reassign the input or/and output of your program to a file, you should also make your input and output in the same way (or use the `SIC.WPR`, `SIC.WPRN` routine). If you mix `READ(5,format)` and `WRITE(6,format)` instructions with `PRINT`, `READ(*,format)` or `WRITE(*,format)` instructions, you might experience some problem. Mixing with C output (`printf` routine for example) may yield to disordered output because of buffering. Use of routine `GAGOUT`, or of the messaging facility, is recommended to printout information.

The monitor uses additional logical units for the `LOG` file, for opening temporary files and for active procedures. These units are taken in the range 50 to 99. Units 50 to 99 are assumed to be available for the routine `SIC_GETLUN`. This subroutine returns the next available logical unit in this range, and keeps track of their usage. Conflict may occur if the user directly opens a file under a unit in this range without allocating the unit through a call to `SIC_GETLUN`. Other units are never used by SIC and other GILDAS programs.

### 1.5.2 Linking on Linux: TO BE UPDATED

On Unix system, the location of the GILDAS software environment is normally defined by executing a simple starting script, like this one, typically in the `.bash_profile` of the user.

```

export GAG_ROOT_DIR=/home/Guilloteau/gildas/gildas-exe-dev
export GAG_EXEC_SYSTEM=pc-cygwin-g95
source $GAG_ROOT_DIR/etc/bash_profile

```

The development environment can be accessed by typing

```
source $GAG_ROOT_DIR$/admin/gildas-env.sh
```

which selects the default compilers to build the system. The development environment architecture uses Makefiles, and the best to develop a new program is just to clone some architecture (see `contrib/example` for a full working example **To be Done**).

## 1.6 The Library Version

Once you have completed the previous operations, an interactive system is available. The system can also be used in batch or within command procedures. Each command handled interactively can also be executed from a program, using either `exec_command` (for simple commands), or `exec_program` (to execute a procedure, or complex commands calling such procedures).

```

call exec_command{command,error}
call exec_program(command)

```

where `command` is a character string containing the command to be executed. `exec_command` is reserved to execute atomic commands, and return an error in case of problems. On the other hand, `exec_program` is typically called with `command= '@ MyProcedure'`. It cannot return an error, but the procedure or code being executed must have its own error handling mechanism.

### 1.6.1 Library Only mode

It is sometimes a burden to have complex languages such as `GREG1\` in your `HELP` when you just want to use the library version of the language. SIC provides a way to load a language in “Library Only” mode. Commands of such languages are accessible only through the `exec_command` mechanism (or an equivalent coding). This mode is useful for building another more elaborate language from one (or more) lower level languages (pay attention to possible recursive programming in doing so however). This mode is called by loading the corresponding language with a negative size specification :

```
SIC_BEGIN(LANG,HELP,-MCOM,VOCAB,VERSION,Dispatch,ErrorRoutine)
```

A language loaded in such a mode can still be accessed interactively or in procedures: the command name must be given with the full language name in it to resolve the command. Similarly, the `HELP` can be accessed by giving the full language and command names.

## 1.7 Using Variables

SIC allows the definition and use of variables. Variables can be declared by the user, or by the program. In the latter case, SIC only remembers the address of the variable, examining its content only when required (command `EXAMINE` for example or any reference to the variable name in a command). This imposes two precautions in the FORTRAN code.

1. The attribute of the SIC variable (`READ-ONLY` or `READ-WRITE`) should be chosen carefully (e.g. a real-time application in which the time is declared as a SIC variable).
2. The corresponding FORTRAN variable must be `SAVED` to prevent the compiler to allocate it on the stack.

### 1.7.1 Definition

There are 5 variable declaration routines following the same calling conventions

Argument Data Type	Name of Subroutine
REAL(KIND=8)	SIC_DEF_DBLE
REAL(KIND=4)	SIC_DEF_REAL
INTEGER(KIND=4)	SIC_DEF_INTE
CHARACTER(LEN=*)	SIC_DEF_CHAR, SIC_DEF_STRN, SIC_DEF_CHARN
LOGICAL(KIND=4)	SIC_DEF_LOGI, SIC_DEF_LOGIN

The calling syntax is the following

```
CALL SIC_DEF_Name (NAME,VARIABLE,READONLY,ERROR)
```

for `LOGI` and `CHAR`, and

```
CALL SIC_DEF_Name (NAME,VARIABLE,NDIM,DIM,READONLY,ERROR)
```

for `DBLE`, `REAL`, `INTE`, `CHARN`, and `LOGIN`, `NDIM` being the number of dimensions,

and DIM the dimensions of the array VARIABLE, and

```
CALL SIC_DEF_STRN (NAME,VARIABLE,LEN,READONLY,ERROR)
```

LEN being the length of VARIABLE.

NAME is the name of the SIC variable, VARIABLE the name of the corresponding FORTRAN variable, READONLY a logical indicating whether the variable should be Read-Only (.TRUE.) or Read-Write (.FALSE.). ERROR is a logical error flag set by SIC if the variable could not be defined.

In addition, SIC\_DEFSTRUCTURE(Name,readonly,error) can be used to define SIC structures.

### 1.7.2 Assignment and Examination

SIC variables can be assigned new values, and examined using the following routines.

Argument Data Type	Assignment	Examination Subroutine
REAL(kind=8)	SIC_LET_DBLE	SIC_GET_DBLE
REAL(KIND=4)	SIC_LET_REAL	SIC_GET_REAL
INTEGER(KIND=4)	SIC_LET_INTE	SIC_GET_INTE
LOGICAL(KIND=4)	SIC_LET_LOGI	SIC_GET_LOGI
CHARACTER(LEN=*)	SIC_LET_CHAR	SIC_GET_CHAR

The calling syntax is the following

```
CALL SIC_LET_Name (NAME,VARIABLE,ERROR)
```

```
CALL SIC_GET_Name (NAME,VALUE,ERROR)
```

except for SIC\_GET\_CHAR

```
CALL SIC_GET_CHAR (NAME,STRING,LENGTH,ERROR)
```

These a-priori useless routines can be used to modify variables defined in completely independent parts of a program such as the codes supporting two different SIC languages. For example the GRAPHIC\ language interacts with GREG1\ and GREG2\ through SIC variables.

For assignment routines, the data type of the VARIABLE argument and of the SIC variable must match exactly. For retrieving routines, the data type to specify is the type of the VALUE argument; implicit conversion from the type of the SIC variable is done if possible, an error is returned otherwise.

Only SCALAR variables can be assigned or examined in this way. For ARRAY variables, you should use the subroutine SIC\_DESCRIPTOR.

```
CALL SIC_DESCRIPTOR (VARIABLE,DESC,FOUND)
```

- VARIABLE is the variable name (general syntax A[i,j] allowed)
- DESC is the descriptor of the variable, an integer array of dimension 9 containing
  - DESC(1) Variable type, with READONLY code. The true variable type is VAR\_TYPE(DESCR)
  - Valid values are
    - \* 0 < VAR\_TYPE(DESC) : Character string of length VAR\_TYPE(DESC)
    - \* VAR\_TYPE(DESC) < 0 may have the value

```

        FMT_R4 for REAL(KIND=4) variable
        FMT_R8 for REAL(KIND=8) variable
        FMT_I4 for INTEGER(KIND=4) variable
        FMT_L  for LOGICAL(KIND=4) variable
        FMT_BY for INTEGER(KIND=1) variable
        FMT_C4 for COMPLEX(KIND=4) variables
    * If VAR_TYPE(DESC) is not equal to DESC(1), the
      variable is READONLY.
DESC(2) Variable address
DESC(3) Number of dimensions
DESC(4) - DESC(7)      Dimensions
DESC(8) Size of allocation (words)
DESC(9) Origin of variable
        0      program defined
        < 0    user defined
        > 0    image
- FOUND is a logical indicating whether the specified variable
  exist or not.

```

The parameters FMT\_R4, FMT\_R8, FMT\_I4, FMT\_L, FMT\_BY and FMT\_C4 are defined in the module gbl\_format.

Three other routines may be required for variable handling:

```
CALL SIC_MATERIALIZE(VARIABLE,DESC,FOUND)
```

is like SIC\_DESCRIPTOR, but allow implicit transposition and sub-arrays in the variable name. An intermediate variable will be created to handle such cases.

SIC\_INCARNATE is used to create an “incarnation” of a variable under a specified type (REAL, DOUBLE, INTEGER).

```
CALL SIC_INCARNATE(FORM,DESC,INCA,ERROR)
```

where

- FORM is the format of the desired incarnation (FMT\_R4, FMT\_I4 or FMT\_R8)
- DESC is the descriptor of the variable
- INCA is the descriptor of the incarnation
- ERROR is a logical error flag

DESC and INCA are Integer arrays of dimension 9. DESC should have been obtained by a previous call to SIC\_DESCRIPTOR.

Once used, the materialization or incarnation may be deleted using routine SIC\_VOLATILE

```
CALL SIC_VOLATILE(INCA)
```

where

- INCA is the descriptor of the incarnation/materialization.

### 1.7.3 Mathematical Formula Handling

Two subroutines are available to decode mathematical or logical expressions, `SIC_MATH` and `SIC_LOGICAL`. In addition, a subroutine is available to decode generalized sexagesimal notation, `SIC_SEXA`. The calling syntax is the following

```
CALL SIC_name (EXPRESSION,LENGTH,VARIABLE,ERROR)
```

where `EXPRESSION` is a character string containing the mathematical or logical expression to be evaluated, `LENGTH` is the number of characters of this expression, `VARIABLE` a `REAL*8` (or `LOGICAL*4`) variable to receive the expression value, and `ERROR` a logical error flag. A generalized sexagesimal notation is for example `A:B:C` or `A:B`, where `A`, `B` and `C` may be variables or expressions or numbers. `B` and `C` values must of course be greater than or equal to 0 and less than 60. The value is returned in units of `A`, the leftmost field.

### 1.7.4 Deleting Variables

Declared variables can be deleted when no longer needed, using routine `SIC_DELVARIABLE`. The calling syntax is the following

```
CALL SIC_DELVARIABLE (NAME,USER,ERROR)
```

where `NAME` is the SIC variable name, `USER` a logical indicating whether program-defined variable are protected against deletion (`.TRUE.`) or can be deleted (`.FALSE.`), and `ERROR` a logical error flag.

## 1.8 Using Functions

It is possible to define user callable functions which are recognized in the mathematical formulae evaluations, using subroutine `SIC_DEF_FUNC` as follows

```
CALL SIC_DEF_FUNC(NAME,SF,DF,NARG,ERROR)
```

where

```
NAME  is the function name
SF    is the address of the single precision implementation of
      the function
DF    is the address of the double precision implementation of
      the function
NARG  is the number of arguments of the function.
ERROR is an error flag
```

Both single and Double precision routines will be called like as (e.g. for two arguments)

```
S = SF(arg1,arg2)      and      D = DF(arg1,arg2)
```

## 1.9 SIC Callable Routines

### 1.9.1 Monitor interface routines

Available subroutines:

SIC_BEGIN	SIC_OPT	SIC_RUN
SIC_INSERT	SIC_LOG	SIC_LANG

Available functions:

SIC_CTRLC	SIC_LIRE	SIC_INTER_STATE
-----------	----------	-----------------

To build a program around the SIC monitor, different routines are provided. Some must always be used, like `SIC_BEGIN`, others are simply provided for additional capabilities. These routines have a standard `SIC_` like name and are fully described in this section.

`SIC_BEGIN (LANG,HELP,NCOM,CCOM,VERSION,Dispatch,Error)`

This subroutine initialize a SIC language and thus is usually the first one called. All arguments are unchanged by the routines and may be passed as immediate values.

- **LANG Character\*(\*)**.  
The language name, as it will appear in the `HELP` and will be returned by `SIC` after command line processing. It is truncated to 12 characters if necessary.
- **HELP Character\*(\*)**.  
The logical name for the `HELP` file corresponding to the language being initialized.
- **NCOM Integer**.  
The number of command and options, i.e. the dimension of the `CCOM` character array
- **CCOM Character\*(\*) array of dimension NCOM**.  
Contains the vocabulary of the language. The structure of the command vocabulary is described elsewhere in this document.
- **VERSION Character\*(\*)**.  
A string to indicate the version number, the last modified date, the programmer name,... which will appear as a message with the language name at run time when using the `SIC VERSION` command
- **Dispatch** The dispatching routine
- **Error** The error handling routine

`SIC_INSERT(LINE)`

This routine is only useful when the Stack Buffer is used to store commands. This is the default, and the case when the flag `MEMORY` was set to `.TRUE.` in the first call to `SIC_OPT`. It can be changed with the `SIC MEMORY` command. `SIC_INSERT` is used to put a command line into the Stack buffer, a very useful possibility for some applications, when several lines must be put or when the command line must be expanded by the user program before the insertion is made.

Note however that command lines are automatically inserted in the Stack Buffer by the monitor for usual commands. This is the standard way of using the Stack insertion mode, since it automatically takes into account many things like the execution level before deciding whether an insertion must occur. Using directly `SIC_INSERT` usually implies to disable the automatic insertion by the monitor (using the character code `*` in the definition of the associated command, c.f. Section 1.2). You need also to perform a call to `SIC_LIRE` to check the `SIC` execution mode (you should only call `SIC_INSERT` when `SIC_LIRE()` is equal to 0) and associated calls to `SIC_LOG` to write the same information in the `LOG` file. The insertion does not occur if the memory flag is turned off (`SIC MEMORY OFF`).

- LINE Character\*(\*) .

SUBROUTINE SIC\_LOG (LINE,NL,LIRE)

This subroutine writes LINE(1:NL) into the Log\_File if LIRE=0. LIRE is here necessary for consistency with the internal pointer indicating if SIC is processing a macro, the stack or the loop. This pointer can be retrieved by the function SIC\_LIRE. A call to SIC\_LOG when LIRE is not 0 has no action.

- LINE Character\*(\*) Input
- NL Integer Input
- LIRE Integer Input

SUBROUTINE SIC\_LANG(LANG)

This subroutine returns the name of the language corresponding to the last command analysed. It is called only when building the library version of a multi-language application.

- LANG Character\*(\*) is returned by the routine

LOGICAL FUNCTION SIC\_CTRL ( )

SIC itself traps the <^C> by generating a PAUSE at the end of the command which was being executed when <^C> was pressed. You may want in time-consuming applications to check yourself at specific points whether <^C> has been pressed. SIC\_CTRL allows you to do so, and resets an internal flag to .FALSE. when called. It returns .TRUE. if <^C> has been pressed since either the last command completed execution or the last time it was called (using the most recent event), .FALSE. otherwise.

INTEGER FUNCTION SIC\_LIRE ( )

This subroutine returns the internal pointer of SIC indicating where SIC is currently reading its commands. SIC\_LIRE may take the values

- -10 Subroutine mode
- -2 Reading in the Loop buffer
- -1 Called from SIC\_RUN with ICODE = -1
- 0 Interactive mode
- I>0 Reading in macro number I

LOGICAL FUNCTION SIC\_INTER.STATE( )

This subroutine returns .TRUE. if the session is interactive, .FALSE. otherwise. This is the case if SIC is ran during a batch process.

## 1.9.2 SIC Arguments Retrieving Routines

Available subroutines:

SIC_L4	SIC_I4	SIC_R4	SIC_R8
SIC_CH	SIC_KE	SIC_NEXT	

Available functions:

SIC_PRESENT	SIC_LEN	SIC_NARG	SIC_START
-------------	---------	----------	-----------

There are 5 retrieving routines following similar calling conventions

Argument Data Type	Name of Subroutine
REAL(KIND=8)	SIC_R8
REAL(KIND=4)	SIC_R4
INTEGER(KIND=4)	SIC_I4
LOGICAL(KIND=4)	SIC_L4
CHARACTER(LEN=*)	SIC_CH, SIC_KE

The calling syntax is the following

```
CALL SIC_Name (LINE,IOPT,IARG,ARGUM,PRESENT,ERROR)
```

Except for SIC\_CH and SIC\_KE

```
CALL SIC_CH (LINE,IOPT,IARG,ARGUM,LARG,PRESENT,ERROR)
```

```
CALL SIC_KE (LINE,IOPT,IARG,ARGUM,LARG,PRESENT,ERROR)
```

- **LINE Character\*(\*)**  
is the last command line processed from which arguments are to be retrieved.
- **IOPT Integer**  
is the number of the option. 0 means the command. Options are numbered in the order where they appear in the vocabulary array after the command name.
- **IARG Integer**  
is the number of the argument of the option/command to be retrieved. 0 means the option/command itself.
- **ARGUM (Type depending on the subroutine called)**  
is the argument to be returned
- **LARG Integer**  
is the returned number of characters for SIC\_CH and SIC\_KE
- **PRESENT Logical**  
is a logical indicating whether the argument must be present (.TRUE.) or not (.FALSE.).
- **ERROR An logical error flag**  
which is set if either i) PRESENT = .TRUE. and the argument is missing or ii) a decoding error occurred.

SIC\_KE is like SIC\_CH, but expects a “keyword”, and thus returns the argument after upper case conversion.

In addition there are other functions related to the arguments:

```
INTEGER FUNCTION SIC_NARG(IOPT)
```

returns the number of arguments present for option IOPT.

```
LOGICAL FUNCTION SIC_PRESENT(IOPT,IARG)
```

returns a logical value indicating the presence of the IARG-th argument of the IOPT-th option.

```
SUBROUTINE SIC_AMBIGS(FACILITY,NAME,FULL,VOCAB,MVOC,IVOC,ERROR)
```

search for NAME in the vocabulary VOCAB, and returns the corresponding pointer IVOC in the vocabulary and corresponding keyword FULL = VOCAB(IVOC) if NAME is a non-ambiguous abbreviation of FULL.



- **FACILITY Character\*(\*)**  
A character string containing the name of the calling subroutine. This string is printed before any failure message (ambiguous or non existing keyword).
- **NAME Character\*(\*)**  
The input name to be searched for in the vocabulary
- **FULL Character\*(\*)**  
The complete keyword returned.
- **IVOC Integer**  
The keyword number in the vocabulary.
- **MVOC Integer**  
The number of keywords in the vocabulary.
- **VOCAB Character\*(\*) (MVOC)**  
The vocabulary.
- **ERROR Logical**  
An error flag set if the input name was ambiguous, invalid or not found in the vocabulary.

Typically, NAME will have been recovered by a call to SIC\_KE

**INTEGER FUNCTION SIC\_LEN(IOPT,IARG)**

returns the total length of the IARG-th argument of the IOPT-th option. Returns 0 if SIC\_PRESENT(IOPT,IARG)=.FALSE.

**INTEGER FUNCTION SIC\_START(IOPT,IARG)**

returns a pointer to the address of the argument in the character string. Returns 0 if SIC\_PRESENT(IOPT,IARG)=.FALSE.. This routine can be used with SIC\_LEN for special processing of arguments, such as sexagesimal decoding.

**SUBROUTINE SIC\_NEXT(LINE(NEXT:),ARGUM,LARG,NEXT)**

This subroutine can be used for special processing of text with a syntax similar to command lines, i.e. text in which the delimiters are single spaces, strings being included between double quotes. Provided it is called with the text restricted to the current delimiter position LINE(NEXT:), it returns the position of the next delimiter NEXT (space, skipping strings), the character string included between these delimiters, ARGUM and its length LARG. This string may further be decoded as double precision value using routine SIC\_MATH or SIC\_SEXA, logical value using routine SIC\_LOGICAL, or used as character constant.

### 1.9.3 Command Line Interpreter Subroutines

**SUBROUTINE SIC\_ANALYSE (COMMAND,LINE,NLINE,ERROR)**

This subroutine analyses the command line LINE, set up all the internal pointers for argument retrieval, and returns the name of the command found. The line must have been formatted by SIC\_FORMAT before being analysed. All active languages are considered, even the “Library Only” languages

- **COMMAND Character(LEN=\*)**, Command name (returned)
- **LINE Character(LEN=\*)**, Line to be analysed. It is modified by SIC\_ANALYSE
- **NLINE Integer**. Current length of LINE, modified on return.

- **ERROR** Logical. Return error code

**SUBROUTINE SIC\_FORMAT(LINE,NLINE)**

This subroutine does an adequate formatting of any line for later processing by the SIC interpreter. It does not use any internal common of SIC and thus may be accessed independently and considered as a more general routine.

- **LINE** Character\*(\*), Line to be formatted.
- **NLINE** Integer, Current length of **LINE**. It is modified by **SIC\_FORMAT** and must be initialized so that only trailing spaces appear in **LINE** after **NLINE**. A good policy is to initialize **NLINE** to **LENC(LINE)**.

#### 1.9.4 All Purpose General Subroutines

Available subroutines:

<b>SIC_LOWER</b>	<b>SIC_PARSEF</b>	<b>SIC_UPPER</b>	<b>SIC_WPR</b>
<b>SIC_WPRN</b>			

Available functions (Obsolescent)

<b>SIC_GETVM</b>	<b>FREE_VM</b>	<b>LENC</b>
------------------	----------------	-------------

These routines are of more general use than SIC itself.

**SUBROUTINE SIC\_LOWER(LINE) - SIC\_UPPER(LINE)**

Converts the string **LINE** to lower or upper case letters respectively.

**SUBROUTINE SIC\_PARSEF (NAME,FILE,DEF,EXT)**

Parses the file name **FILE** for a default directory **DEF** and extension type **EXT**. These item are added to **FILE** if needed, and **NAME** returns the short **NAME** of the file, that is, the name without extension or directory. All arguments are Character\*(\*). This is the standard way to obtain a file name in the GILDAS environment: this routine takes care of logical names (in the SIC meaning).

**SUBROUTINE SIC\_WPR(PROMPT,LINE)**

Reads the string **LINE** with the prompt **PROMPT**. It tries to obtain a non-blank string, prompting again if it reads a blank string. If it receives a <^Z> code, it returns the string "EXIT". The prompt is automatically disabled if the session is not interactive.

- **PROMPT** Character\*(\*), Unchanged by **SIC\_WPR**
- **LINE** Character\*(\*), Returned line.

**SUBROUTINE SIC\_WPRN(PROMPT,LINE,N)**

Reads the string **LINE** with the prompt **PROMPT**. Returns **N** the number of characters read. Returns **N=0** if it reads a blank line, or receives a <^Z> code. The prompt is automatically disabled if the session is not interactive.

- **PROMPT** Character\*(\*), Unchanged by **SIC\_WPRN**
- **LINE** Character\*(\*), Returned line.
- **N** Integer, Number of characters in **LINE**

### 1.9.5 Symbol Manipulation Routines

Three routines are available to define, translate and delete symbols within a program using SIC without using the `SYMBOL` command.

`SUBROUTINE SIC_SETSYMBOL(SYMBOL,TRANSLATION,ERROR)`

Defines a symbol of name `SYMBOL` (Character\*(\*), maximum 12 characters), translation `TRANSLATION` (Character\*(\*), maximum 132 characters). An error flag (`ERROR` logical) is returned if the symbol could not be defined. Any previous definition is overridden.

`SUBROUTINE SIC_GETSYMBOL(SYMBOL,TRANSLATION,ERROR)`

Obtains the current translation of a given symbol. If `TRANSLATION` is too short, the translation is truncated without warning or error. `ERROR` is returned if the symbol is undefined only.

`SUBROUTINE SIC_DELSYMBOL(SYMBOL,ERROR)`

Deletes a symbol definition. `ERROR` is returned if the symbol was not defined.

## 1.10 Obsolescent Features

### 1.10.1 Function

`INTEGER FUNCTION LENC(LINE)`

Returns the “current” length of the string `LINE`, that is the position of the last significant character. It thus allows to ignore trailing blanks. It is obsolescent, as the Fortran-90 intrinsic `LEN_TRIM` provides the same functionality.

`INTEGER FUNCTION SIC_GETVM(N32,ADDR)`

`SIC_GETVM` allocates virtual memory for `N32` 32-bit words, and returns the allocated address space into the `KIND=ADDRESS_LENGTH` variable `ADDR`. `SIC_GETVM` is set to 1 in case of success, to other values otherwise.

`SUBROUTINE FREE_VM(N32,ADDR)`

`FREE_VM` can be used to free the corresponding address space when no longer needed.

`SIC_GETVM` and `FREE_VM` use is being deprecated, as the `ALLOCATE/DEALLOCATE` mechanism of Fortran-90 offers an equivalent functionality.

### 1.10.2 Routines for Library usage (Obsolescent, for record only)

`SIC_OPT (PROMPT,LOGFILE,MEMORY)`

This routine is not compulsory. It is used to set the prompt the log file name and the stack usage. This routine only changes the prompt if called after `SIC_BEGIN`.

- `PROMPT Character*(*)`.

The prompt to be used in interactive mode, truncated to 8 characters if necessary. Default is `SIC`. Note that the caret `>` and other alterations such as `:` or `_1>` are added by `SIC` at run time and should not be included.

- `LOGFILE Character*(*)`.

The log file name. Default is `LOG`.

- `MEMORY Logical`.

Indicates whether the Stack is used or not.

`SIC_RUN (LINE,LANG,COMMAND,ERROR,ICODE,OCODE)`

This subroutine is used to enter SIC, retrieve and analyse a command line for further execution. All SIC possibilities, including execution levels and error recoveries, made accessible (In particular, the LINE command line can be a call to a macro file).

- **LINE Character(\*)** (maximum size is 2048)  
Command line to be executed. Modified by SIC. No need to initialize it when ICODE = 0 or 1, but must be initialized is ICODE = -1 or 2
- **LANG Character\*12**  
Name of the language returned by SIC
- **COMMAND Character\*12**  
Name of the command returned by SIC
- **ERROR Logical**  
Return error code
- **ICODE Operation code**
  - ICODE = -1  
Analyse the command line passed as argument LINE, and return to calling program to execute it.
  - ICODE = 0  
Loop into SIC to retrieve a new command line LINE, and return to calling program to execute it.
  - ICODE = 1  
Start SIC, and retrieve a first command line.
  - ICODE = 2  
Start SIC, analyse the command line passed as argument, and return to calling program to execute it.
- **OCODE Return code**
  - OCODE = -1  
End of execution in sub-routine mode. Program must return immediately to its caller.
  - OCODE = 0  
Successful analysis of a command. Program must execute it, and then loop again on SIC\_RUN with ICODE=0 to get further commands.
  - OCODE = 1  
End of execution caused by typing the EXIT command. Program should perform any necessary action (close files, etc...) and return to its caller.

## Chapter 2

# GreG Programming Manual

**Partially updated in Oct 2008. Need more details about loading GreG.**

This section describes how GREG normally used as an interactive plot utility, can also be used as a high level plot library. GREG can be used exactly as a standard graphic library, but because of the possibilities of the command line monitor, many other possibilities are accessible.

Before presenting in detail the “Library Version” of GREG we should distinguish between three different possible applications of the GREG Library :

1. The occasional user who has a single repetitive graphic problem which is part of another complex program. Interactive control is not wanted. This case can often be solved using command procedures and SIC images as data format when formatted I/O is definitely too slow. This is very efficient and flexible. If the user already discarded this possibility, he (she) most likely wants the simplest programming ever possible, will be satisfied by standard default values, and is not really worried about optimum efficiency.
2. A programmer wanting to solve a single repetitive graphic problem for use by other people. Interactive action is not wanted. Simple programming is of little importance, but efficiency is a major problem.
3. A programmer wishing to integrate elaborate and flexible graphic applications as part of a more complete data analysis system. Interactive control by the user and error recovery are necessary. Then all GREG capabilities are wanted, and in addition this programmer may well be interested to use the possibilities of the SIC monitor to “supervise” the data analysis system.

The library version of GREG allows all three cases to be solved adequately by offering three different ways to call GREG services :

1. Passing a command to GREG using the routine `GR_EXEC`  
`GR_EXEC('Command Argument/Option')` will execute the command line exactly as if you had typed it interactively. Instead of `GR_EXEC`, it is recommended to use `GR_EXEC1`, `GR_EXEC2` and `GR_EXECL` for commands of languages GREG1, GREG2 and GTVL respectively.
2. Calling an intermediate formatting routine which generates the appropriate call to `GR_EXEC` from its own arguments. Special entries are used to process possible options. The advantage of this mode is to provide a more standard program interface. Not all commands will be accessible in that way, and it is marginally slower than the previous mode.

3. Calling subroutines which do not correspond to GREG commands but directly generates plot actions. This way is the most optimized access, since it bypasses the command line interpreter. However, only standard things can be done like this, and it requires some precautions because of the segmented nature of the graphic library.

All three ways can be used within a single program, and the choice between one or the other is just a matter of convenience and/or efficiency.

In addition, the GREG plot library can be used either as a classic package of subroutines (“Library Only”), or as an interactive facility allowing user control at run time by means of the SIC monitor possibilities (“Interactive”). The two modes can be mixed in a subroutine, with the important restriction that the “Library Only” mode is just a subset of the “Interactive” mode. It is not possible to change the mode during program execution. A program using GREG plot library in “Interactive” mode in fact appears as a superset of the GREG interactive utility.

## 2.1 Interaction with FORTRAN programs

For a normal user, it can be said that GREG does not interfere at all with a FORTRAN program. All interactions with a program concern the system or SIC facilities as detailed below.

- I/O Units :  
All logical units used by the GREG system and the associated SIC monitor are FORTRAN units between 50 and 99. Programs using SIC should avoid opening such units, or should get available units through calls to SIC\_GETLUN.
- Work Space :  
GREG uses as much as possible the concept of virtual memory. This means that work space, when required, is allocated dynamically at run time. Hence, GREG does not overload a small program. There is currently one exception, the X,Y and Z buffers which have a fixed size allocation of 10000 long-words each. This may change at some time. On large applications, be sure that your virtual memory quota is large enough. Be sure also to run GREG in a large enough working set to reduce page faults.
- Special Handler :  
The SIC monitor always traps the ^C action to provide a facility to interrupt procedures at any time. You can bypass this action by adequate programming (see SIC\_CTRLC routine in the SIC manual).

Programmers using SIC as command monitor together with GREG either in interactive or in library mode, should be aware of the interaction between GREG and SIC command parsing facility. Each call to GR\_EXEC, GR\_EXEC1, GR\_EXEC2, GR\_EXECL or EXEC\_GREG parses at least one GREG command line, and thus modify the pointers accessed by SIC argument retrieving routines. Accordingly, any subroutine implementing a user command should retrieve all its arguments **before** calling a GREG subroutine.

## 2.2 Basic Routines

### 1. LOAD\_GREG (Mode)

**To be updated since the change of SIC initialization mode** This subroutine must

be called before any other reference to GREG services. It is used to define the operating mode and to initialize the GREG vocabulary into the SIC monitor. The argument is of type Character and can be 'LIBRARY' to initialize GREG in the "Library Only" (no interactive SIC monitor) mode, or 'INTERACTIVE' to initialize GREG with all SIC monitor capabilities. Strings like 'LIBRARY GREG1' or 'INTERACTIVE GREG2' can be used to load only one language, although the GREG2\ language requires GREG1\ to function. This routine does not set the SIC prompt, neither the Log File which you may specify by calling routine SIC\_OPT. Please refer to the SIC manual.

#### 2. GR\_EXEC(LINE)

This is the basic routine for all plot actions. It is able to activate any GREG command exactly in the same way as if you had typed it during an interactive session. Use GR\_EXEC if you do not know to which of the GREG languages it pertains, GR\_EXEC1,2 or L according to the language if you know (and you should). The command line must not include the language field. The command is not written to the SIC stack, neither to the Log File.

#### 3. GR\_ERROR()

This logical function allows error recovery. It returns the internal error status of GREG program, and clears it. If another GREG subroutine is called while an error status exist, program execution aborts.

#### 4. EXEC\_GREG(LINE)

**Obsolescent, use EXEC\_COMMAND or EXEC\_PROGRAM instead** execute a command line which can be either a GREG command or a SIC command (like '@ PROC' for example). Control returns to the calling program when the command is completed successfully. If an error occurs, the current SIC error recovery action is activated before. This is usually a PAUSE which gives interactive control to the user ; control will then return to the calling program only after command EXIT (or ^Z) has been typed. Contrary to GR\_EXEC, the command is written to the SIC stack, and to the Log File. "Interactive" mode is required.

*Only GR\_EXEC (or its variant) and GR\_ERROR are allowed in Library Mode.*

## 2.3 Linking

### 2.3.1 UNIX systems

#### To be updated with current information on architecture under Linux

To access GREG from your Fortran Program, you need to link to several libraries. All GREG libraries are located in \$GAG\_ROOT/lib where \$GAG\_ROOT is an environment variable defining where the GILDAS software is located. See your local GILDAS expert for that.

Then you should use the following link command to link your program

```
f77 -o Program Program.f -L/usr/lib/X11 -L$GAG_ROOT/lib \
-lgdf -lGREG -lcontour -lgtlang -lchar -lsic -limage -lgag -lrar \
-lX11 -lm -lc
```

where

```
f77          is the Fortran compiler/linker
```

<code>Program.f</code>	is your Fortran program.
<code>/usr/lib/X11</code>	indicates where the X11 libraries are located
<code>-lm -lc</code>	are required on some systems

On some systems, the GREG libraries are shareable, thus avoiding code to be included in your own programs.

## 2.4 Running

If you are using GREG in “Library Only” mode, you can just run your program normally.

If you are using the “Interactive” mode you need to provide the assignments of GREG help files and working files to your program. All compulsory assignments are made system wide, so you can just run the program. You will obtain help for the SIC command monitor and a log file will be written in `.gag/logs`.

## 2.5 Example

**No simple working example is available as of today.** See the MAPPING program for a more exhaustive example.

## 2.6 Array Transfer

Subroutines with data arrays transferred to or obtained from GREG have a Real\*4 and a Real\*8 version. The single precision has a name beginning by `GR4_`, the double precision a name beginning by `GR8_`. In the argument list description, `REAL` will mean `Real(KIND=4)` for the `GR4_` version, `Real(KIND=8)` for the `GR8_`. These routine work by copying the data. For really big arrays, it is more efficient to transfer the information by reference, though SIC variables (see the SIC manual).

### 2.6.1 GR4\_GIVE - GR8\_GIVE

`GR4_GIVE(NAME,NXY,ARRAY)`

This subroutine passes One-Dimensional array `ARRAY` to GREG as the X, Y, or Z array according to the given `NAME`. It is the optimal way to initialize the One-Dimensional arrays of GREG with the data you have computed within your application program.

<code>NAME</code>	is CHARACTER(LEN=1) and may be X, Y or Z
<code>NXY</code>	is the number of values set in <code>ARRAY</code>
<code>ARRAY</code>	is a REAL array of size <code>NXY</code> .

### 2.6.2 GR4\_GET - GR8\_GET

`GR4_GET(NAME,NXY,ARRAY)`

This subroutine is the reverse of the `GR4(8)_GIVE` routines. It passes One-Dimensional array `ARRAY` from GREG to your program as the X, Y, or Z array according to the given `NAME`. It allows to benefit in your program of the flexible input formats of GREG.



NAME	is CHARACTER(LEN=1) and may be X, Y or Z	Input
NX	dimension of ARRAY	Input
NY	number of values returned in ARRAY	Output
ARRAY	is a REAL array of size NXY	Output

### 2.6.3 GR4\_RGIVE - GR8\_RGIVE

GR4\_RGIVE(NX,NY,CONV,R)

This subroutine passes a two-Dimensional array to GREG as the Regular Grid array used for mapping.

NX,NY	Integers, array dimensions
CONV	Real(KIND=8) array of dimension 6 It contains respectively CONV(1) Reference pixel in X (First dimension of R) CONV(2) X User coordinate value at CONV(1) CONV(3) X User coordinate increment per pixel (may be negative) CONV(4)-CONV(6) same as above for Y (Second dimension of R)
R	REAL array (NX,NY) to be transferred.

If R is Real(KIND=4), the array is not physically copied but its address is computed by GREG for later use. GREG will not modify anything in it. If of type Real(KIND=8), virtual memory is allocated to create a Real(KIND=4) array of same dimensions.

### 2.6.4 GR4\_LEVELS - GR8\_LEVELS

GR4\_LEVELS(NL,LEVELS)

This subroutine initializes a set of contour levels for mapping by GREG. LEVELS is a REAL array of dimension NL.

## 2.7 Immediate Routines

The following routines do not use the SIC command line interpreter. They are essentially internal GREG routines made available to the user, and are thus very efficient to use. As above, they have a KIND=4 and a KIND=8 version.

For optimization purposes, these routines do not include any explicit segmentation of the plot. In order to be consistent with the philosophy of the interactions between GREG and the GTVIRT graphic library, you must enclose a set of calls to the immediate routines between a call to GR\_SEGM and GR\_OUT. Ex:

```
CALL GR_SEGM('Nom',ERROR) ! Close current graphic segment and open
                           ! a new one
CALL GR4_CONNECT (...)    ! Fill this segment with plot coordinates
CALL GR4_MARKER (...)     ! etc..
CALL GR_OUT               ! Make the segment visible on screen
```

### 2.7.1 GR\_SEGM

GR\_SEGM(NAME,ERROR

Close the current graphic segment and open a new graphic segment with the current plotting attributes selected by PENCIL. All the commands of GREG open at least one new segment, but the immediate routines do not. This routine must be called prior to calling a series of immediate routines to ensure that the plot will appear with the currently selected graphic attributes, and to allow a selective erasure of the plot. NAME is a character string indicating the desired name of the segment, and ERROR a logical error flag. All plot request issued between two successive calls to GR\_SEGM will make a single graphic segment. Note however that every GREG command having an effective plot action creates one or more graphic segments.

### 2.7.2 GR\_OUT

GR\_OUT

Updates the graphic output. This routine must be called when you wish to make visible a series of calls to the immediate routines.

### 2.7.3 DRAW - RELOCATE

DRAW(XU,YU) RELOCATE(XU,YU)

Basic pen down or up movement in User coordinates (Real(KIND=8) values). There is no Real(KIND=4) version.

### 2.7.4 GDRAW - GRELOCATE

GDRAW(X4,Y4) GRELOCATE(X4,Y4)

Basic pen down or up movement in Physical coordinates (Real(KIND=4) values). There is no Real(KIND=8) version.

### 2.7.5 GR4\_PHYS\_USER - GR8\_PHYS\_USER

GR4\_PHYS\_USER (XP,YP,XU,YU,NXY)

Convert Physical coordinates (XP,YP always Real(KIND=4) values) to User coordinates (XU,YU, REAL values). NXY is the number of values.

### 2.7.6 GR4\_USER\_PHYS - GR8\_USER\_PHYS

GR4\_USER\_PHYS(XU,YU,XP,YP,NXY)

Convert User coordinates (XU,YU, REAL values) to Physical coordinates (XP,YP, always Real(KIND=4) values). NXY is the number of values.

### 2.7.7 GR4\_CONNECT - GR8\_CONNECT

GR4\_CONNECT(NXY,X,Y,BVAL,EVAL)

This subroutine connects all data points represented by the X and Y arrays passed in arguments. BVAL and EVAL are used for blanked values. EVAL negative means no blanking.

Arguments :

NXY	INTEGER	Input
-----	---------	-------

X	REAL (NXY)	Input
Y	REAL (NXY)	Input
BVAL	REAL	Input
EVAL	REAL	Input

### 2.7.8 GR4\_HISTO - GR8\_HISTO

GR4\_HISTO(NXY,X,Y,BVAL,EVAL)

Arguments :

NXY	INTEGER	Input
X	REAL (NXY)	Input
Y	REAL (NXY)	Input
BVAL	REAL	Input
EVAL	REAL	Input

### 2.7.9 GR4\_MARKER - GR8\_MARKER

GR4\_MARKER(NXY,X,Y,BVAL,EVAL)

Markers of current size are plotted at each data point.

Arguments :

NXY	INTEGER	Input
X	REAL (NXY)	Input
Y	REAL (NXY)	Input
BVAL	REAL	Input
EVAL	REAL	Input

### 2.7.10 GR4\_CURVE - GR8\_CURVE

GR4\_CURVE(NXY,X,Y,Z,VAR,PER,BVAL,EVAL,ERROR)

Plots a smooth curve from the (X,Y) values using the requested interpolant. Z is either a dummy argument or a parameter for the curve representation depending on the VAR value. VAR indicates which is the variable to use for interpolation. PER indicates whether the curve is periodic or not. BVAL and EVAL define the blanking value and the blanking tolerance (set EVAL negative to disable blanking checking). ERROR is an error flag set if the curve could not be produced. The current accuracy is used for the interpolation.

Arguments :

NXY	INTEGER	Input
X	REAL (NXY)	Input
Y	REAL (NXY)	Input
VAR	CHARACTER*(*)	Input
PER	LOGICAL	Input
BVAL	REAL	Input
EVAL	REAL	Input
ERROR	LOGICAL	Output

### 2.7.11 GR4\_EXTREMA - GR8\_EXTREMA

GR4\_EXTREMA(NXY,X,BVAL,EVAL,XMIN,XMAX,NMIN,NMAX)

Compute the extrema of the input array avoiding blanked values.

Arguments :

NXY	INTEGER	Input	Number of points
X	REAL (NXY)	Input	Array
BVAL	REAL	Input	Blanking value
EVAL	REAL	Input	Tolerance on blanking
XMIN	REAL	Output	Minimum value
XMAX	REAL	Output	Maximum value
NMIN	INTEGER	Output	Pixel of the minimum X(NMIN) = XMIN
NMAX	INTEGER	Output	Pixel of the maximum

### 2.7.12 GR8\_BLANKING

GR8\_BLANKING(BVAL,EVAL)

Define the blanking value to be used later. It is equivalent to call GR\_EXEC('SET BLANKING BVAL EVAL'), but the later form requires formatting of values.

Arguments :

BVAL	REAL	Input	Blanking value
EVAL	REAL	Input	Blanking precision

### 2.7.13 GR8\_SYSTEM - GR8\_PROJEC

GR8\_SYSTEM (ICODE)

GR8\_PROJEC (X,Y,A,ICODE)

Define respectively the coordinate projection system and the projection constants. The SYSTEM code can be 1 for UNKNOWN, 2 for EQUATORIAL (1950.0), 3 for GALACTIC. The projection code is 0 for NONE, 1 for GNOMONIC, 2 for ORTHOGRAPHIC, 3 for AZIMUTHAL, 4 for STEREOGRAPHIC, 5 for LAMBERT cylindrical, 6 for AITOFF equal area and 7 for RADIO (also known as global SINUSOIDAL) projection.

Arguments :

ICODE	INTEGER	Input	
X	REAL	Input	Projection center
Y	REAL	Input	Projection center
A	REAL	Input	Projection Angle

### 2.7.14 GR4\_RVAL

GR4\_RVAL(XU,YU,Z4)

Returns a map value at a given point in user coordinates.

Arguments :

XU	REAL*8	Input
YU	REAL*8	Input
Z4	REAL*4	Output

**2.7.15 GR\_WHERE**

GR\_WHERE(XU,YU,X4,Y4)

Returns the pen position with the same conventions as GR\_CURS

Arguments :

XU	R*8	X User coordinates	Output
YU	R*8	Y User coordinates	Output
X4	R*4	X Plot coordinates	Output
Y4	R*4	Y Plot coordinates	Output

**2.7.16 GR8\_TRI**

GR8\_TRI(X,INDEX,N,\*)

Sorting program that uses a quicksort algorithm. Applies for an input array of Real(KIND=8) values which is reordered. It also returns an array of indexes sorted for increasing order of X. You can use GR8\_SORT to reorder other arrays associated with X.

Arguments :

X	R*8(*)	Unsorted/Sorted array	Input/Output
INDEX	I(*)	Integer array of sorted indexes	Output
N	I	Length of arrays	Input
*	Label	Error return	

**2.7.17 GR8\_SORT**

GR8\_SORT(X,XSORT,INDEX,N)

Reorder a real\*8 array using the sort indexes computed by GR8\_TRI. Note that X and XSORT must be different (i.e. sorting cannot take place within the same array).

Arguments :

X	R*8(*)	Unsorted/Sorted array	Input/Output
XSORT	R*8(*)	Sorted array	Work space
INDEX	I(*)	Integer array of sorted indexes (obtained by GR8_TRI)	Input
N	I	Length of arrays	Input

**2.7.18 GR\_CLIP**

LOGICAL FUNCTION GR\_CLIP(clip)

Turn on (CLIP = .TRUE.) or off (CLIP = .FALSE.) clipping of lines, and return current status in GR\_CLIP. By default, clipping is on. Caution : some GREG subroutines force clipping off and reset it on upon exit.

**2.8 The cursor routine**

GR\_CURS (XU,YU,X4,Y4,CODE)

Calls the interactive graphic cursor and returns its position when you hit any alphanumeric key on the keyboard.

Arguments :

XU	REAL*8	X User coordinates	Output
YU	REAL*8	Y User coordinates	Output
X4	REAL*4	X Plot coordinates	Output
Y4	REAL*4	Y Plot coordinates	Output
CODE	CHARACTER*1	Character struck	Output

The subroutine returns CODE = 'E' when an error occurs.

For X-Window terminals, button 1 returns ^ , 2 returns & and 3 returns \*.

## 2.9 GREG High-Level Subroutines

These routines essentially format the command line to pass it later to the GR\_EXEC subroutine, and are thus the less efficient routines of the GREG library. They are provided essentially for user convenience because the formatting needed to use GR\_EXEC might be tedious. Note all GREG commands have the high level equivalent, but all can be used with GR\_EXEC.

Each subroutine corresponds to a command, and each entry corresponds to an option of that command. The subroutine and entry names are built from the 4 first characters of the corresponding command and option names. The entries must be called before the subroutine, since it is this one which effectively transmit the command to GR\_EXEC.

The conventional type of arguments are

NAME	Character
ARG, ARG1, ...	Real*8
IARG, IARG1, ...	Integer*4
NARG	Integer*4

When NARG is present, all arguments need not be passed : trailing arguments may be omitted. NARG is used to give the actual number of arguments passed after NARG itself and it must be set precisely. All arguments before NARG, usually the character string NAME, and the argument NARG must always be present (even if NAME=' '). Note that if ARG1 is missing, ARG2 cannot be present. Note also that the double quotes should not be passed for character strings. For instance, the interactive command LABEL 'A little toy' should be replaced by CALL GR\_LABE('A little toy').

Example : CALL GR\_DRAW ('MARKER', 2, 0.3D0, 0.8D0)

Note that the ARG1 and ARG2 arguments are here expressed as constants, and the D exponent is required as they must be Real\*8 values.

The high level routines are not yet fully guaranteed as up to date. In the following description, each high level subroutine is preceded by the assumed syntax for the equivalent command. If this syntax differs from the current internal help description, using the discrepant parts of the high level routine will cause a fatal error, but common parts of the syntax can be used safely.

For the GTVL language, the following routines are available

CLEAR [Argument]

SUBROUTINE GR\_CLEA(NAME)

DEVICE [Type [Descriptor]] [/OUTPUT Logical\_Name]

SUBROUTINE GR\_DEVI(NAME)

All arguments of command DEVICE being character strings, it is easy to concatenate them into NAME.

For the GREG1 language, the routines are

```

AXIS Name [A1 A2] [/LOCATION X Y] [/TICK Orientation [BIG SMALL]]
      [/LABEL Position] [/[NO]LOG] [/ABSOLUTE]
      SUBROUTINE GR_AXIS(NAME,NARG,ARG1,ARG2)          ! 28-Sep-1986
      ENTRY GR_AXIS_LOCA(NARG,ARG1,ARG2)
      ENTRY GR_AXIS_TICK(NAME,NARG,ARG1,ARG2)
      ENTRY GR_AXIS_LABE(NAME)
      ENTRY GR_AXIS_LOG
      ENTRY GR_AXIS_NOLO
      ENTRY GR_AXIS_ABSO

```

```

BOX [Arguments]
      SUBROUTINE GR_BOX(NAME)                          ! 28-Sep-1986
      ENTRY GR_BOX_ABSO

```

```

COLUMN [X Nx] [Y Ny] [E Ne] [/FILE File] [/LINES Lmin [Lmax]]
      [/TABLE TableName]
      SUBROUTINE GR_COLU(NAME)                          ! 28-Sep-1986
      As this routine is normally of no use in the library version,
      it requires you to code explicitly the remaining part of the
      command line. Hence, it should be equivalent to use
      either GR_EXEC('GREG1\COLUMN '//NAME)
      or GR_EXEC1('COLUMN '//NAME)

```

```

CONNECT [/BLANKING Bval Eval]                          ! 28-Sep-1986
      SUBROUTINE GR_CONN
      ENTRY GR_CONN_BLAN(NARG,ARG1,ARG2)

```

```

DRAW [Action [X Y]] [/USER] [/BOX N] [/CHARACTER N] [/CLIP]
      SUBROUTINE GR_DRAW(NAME,NARG,ARG1,ARG2)
      ENTRY GR_DRAW_CHAR(NARG,IARG)
      ENTRY GR_DRAW_USER
      ENTRY GR_DRAW_BOX(NARG,IARG)
      ENTRY GR_DRAW_CLIP

```

Note that no support is given for DRAW TEXT and DRAW FILL\_AREA, unless you specify in NAME all the arguments and set NARG to 0. DRAW TEXT can easily be replaced by GR\_DRAW('RELOCATE'...) followed by a call to GR\_LABE.

```

ERRORBAR NAME
      SUBROUTINE GR_ERRO(NAME)                          ! 28-Sep-1986

```

```

HISTOGRAM [/BASE [Ybase]] [/BLANKING Bval Eval]
      SUBROUTINE GR_HIST                                ! 28-Sep-1986
      ENTRY GR_HIST_BASE(ARG)

```

```

ENTRY GR_HIST_BLAN(NARG,ARG1,ARG2)

LABEL "String" [/X] [/Y] [CENTERING N] [/APPEND]
  SUBROUTINE GR_LABE(NAME)                                ! 28-Sep-1986
  ENTRY GR_LABE_X
  ENTRY GR_LABE_Y
  ENTRY GR_LABE_CENT(IARG)
  ENTRY GR_LABE_APPE

LIMITS [Xmin Xmax Ymin Ymax] [/XLOG] [/YLOG] [/RGDATA] [/REVERSE [X] [Y]]
  [/BLANKING Bval Eval]
  SUBROUTINE GR_LIMI(NARG,ARG1,ARG2,ARG3,ARG4)            ! 28-Sep-1986
  ENTRY GR_LIMI_XLOG
  ENTRY GR_LIMI_YLOG
  ENTRY GR_LIMI_RGDA
  ENTRY GR_LIMI_REVE(NAME)
  ENTRY GR_LIMI_BLAN(NARG,ARG1,ARG2)
  Arguments of GR_LIMI can be omitted with the standard VAX
  convention to indicate that automatic limit must be computed
  for this one. For example
      CALL GR_LIMI_XLOG
      CALL GR_LIMI(4,0.d0, ,-10.d0,10.d0)
  is equivalent to
      CALL GR_EXEC1('LIMITS 0 * -10 10/XLOG')
  Presently, there is no support for the fifth argument (Angular
  unit or Absolute coordinates) and for the =, < and > possibilities
  of command LIMITS in this High Level routine. To use these
  possibilities, you should use directly GR_EXEC1('LIMITS ...').

PENCIL [N] [/COLOUR C] [/DASHED D] [/WEIGHT W] [/DEFAULT]
  SUBROUTINE GR_PEN(NARG,IARG)                            ! 28-Sep-1986
  ENTRY GR_PEN_DASH(IARG)
  ENTRY GR_PEN_WEIG(IARG)
  ENTRY GR_PEN_COLO(IARG)
  ENTRY GR_PEN_DEFA

POINTS [Size] [/BLANKING Bval Eval]
  SUBROUTINE GR_POIN(NARG,ARG1)
  SUBROUTINE GR_POIN_BLAN(NARG,ARG1,ARG2)

RULE [X] [Y] [/MAJOR [/MINOR]
  SUBROUTINE GR_RULE(NAME)                                ! 28-Sep-1986
  ENTRY GR_RULE_MAJO
  ENTRY GR_RULE_MINO

SET Something [Value1 [Value2]...]]
  SUBROUTINE GR_SET(NAME,NARG,ARG1,ARG2,ARG3,ARG4) ! 28-Sep-1986

```



SHOW

```
SUBROUTINE GR_SHOW(NAME) ! 28-Sep-1986
```

TICKSPACE SmallX BigX SmallY BigY ! 28-Sep-1986

```
SUBROUTINE GR_TICK(NARG,ARG1,ARG2,ARG3,ARG4)
```

The following commands correspond to the GREG2\ Language.

EXTREMA [/BLANKING Bval Eval] [/PLOT]

```
SUBROUTINE GR_EXTR ! 28-Sep-1986
```

```
ENTRY GR_EXTR_BLAN(NARG,ARG1,ARG2)
```

```
ENTRY GR_EXTR_PLOT
```

LEVELS List

```
SUBROUTINE GR_LEVE(NAME)
```

Coding the list in the required format may be funny.

Use GR4\_LEVELS or GR8\_LEVELS instead, they are more convenient.

RGDATA File\_Name [/SUBSET IX1 IY1 IX2 IY2]

```
SUBROUTINE GR_RGDA(NAME) ! 28-Sep-1986
```

```
ENTRY GR_RGDA_SUBS(NARG,IARG1,IARG2,IARG3,IARG4)
```

RGMAP [/ABSOLUTE Value] [/PERCENT Value] [/KEEP] [/BLANKING Bval Eval]

[/GREY Colour Ntry] [/PENS Pos Neg]

```
SUBROUTINE GR_RGMA ! 28-Sep-1986
```

```
ENTRY GR_RGMA_ABSO(ARG)
```

```
ENTRY GR_RGMA_PERC(ARG)
```

```
ENTRY GR_RGMA_BLAN(NARG,ARG1,ARG2)
```

```
ENTRY GR_RGMA_KEEP
```

```
ENTRY GR_RGMA_PENS(NARG,IARG1,IARG2)
```

No support is yet given for option /GREY because it is still rather experimental.

Remember that if anything available in interactive seems to be missing in the previous list, you can always use GR.EXEC to access it.



## Chapter 3

# Task Programming Manual

### 3.1 General Outline and Data Structure of Images

The GILDAS software is designed to work in an heterogeneous network, where each computer may have its own floating point and integer number representation and its own operating system. Images created on any computer should be accessible transparently from any other computer in the network.

To allow such a portability and yet preserve efficiency when working on a single type of machines, all files are written in the binary representation of the machine on which they were created. A library of subroutines is used to access the images and perform the necessary format conversion between the different representations of real and integer numbers. So far, 3 representations are recognised: VAX format, IEEE format (DEC DS Workstation) and IEEE format (Sun SparcStation, IBM RS-6000 series, etc...).

The images have the following organisation :

- Images are basically direct access files with a blocksize of 512 bytes. This correspond to the system block size on VAX-VMS machines.
- The first block is a header block, defining the size of the image and all major parameters, such as World Coordinate System definition.
- The following blocks containing the image array itself.
- There may be trailing blocks for additional information. These trailing blocks are not compulsory, and may be ignored by the processing software.

The header starts with 'GILDAS', followed by one character which indicates the type of integer and floating point number representation

- Underscore “\_” for VAX machines
- Dot “.” for swapped IEEE machines (IBM RS-6000, SUN Sparc, etc...)
- Minus “-” for non-swapped IEEE machines (DEC DS series)

and extra text indicating the type of information in the image:

- IMAGE for all images (up to 4-D arrays)
- UVDAT for tables of unsorted, unprocessed UV data

- UVSOR for tables of sorted, preprocessed UV data

On computers which support such possibilities (VMS systems), GILDAS makes extensive use of memory mapping to access the images : instead of being read into memory through standard I/O techniques, the disk file containing an image is directly made part of the virtual memory accessed by the program, through the use of the appropriate system routines. Physical I/O between this virtual memory and the physical memory is then later handled by the system memory management.

On other computers, the image is copied into memory using standard FORTRAN direct I/O. Provided adequate programming by the application programmer, only the necessary part of the image is read or mapped. This is a totally transparent process for the application programmer, who will always consider the image as a FORTRAN array in virtual memory.

The portable version of the GILDAS software makes the distinction between an image and the "incarnation" of an image subset in memory. Each image has an associated "Image Slot" (IS), while each memory part has an associated "Memory Slot" (MS). Several memory slots may corresponds to several "windows" into a single image slot.

## 3.2 Fortran-90 access to images

Fortran-90 allows definition of data structures (derived types) which are convenient to handle images. The Fortran-90 module `image_def` defines two derived type corresponding to GILDAS images: the type GILDAS and the type SIC.

```
!
module image_def
  use gildas_def
  !
  integer, parameter :: code_read_data      ==-1
  integer, parameter :: code_read_header    ==-2
  integer, parameter :: code_update_header  ==-3
  integer, parameter :: code_write_header   ==-4
  integer, parameter :: code_write_data     ==-5
  integer, parameter :: code_read_image     ==-6
  integer, parameter :: code_free_image     ==-7
  integer, parameter :: code_create_image   ==-8
  !
  ! /XIPAR/
  type :: location
    sequence
    integer(kind=4) :: al64                ! Padding to align ADDR
    integer(kind=4) :: size                ! Size of image
    integer(kind=address_length) :: addr   ! Address of image
    integer :: islo                        ! Image Slot number
    integer :: mslo                        ! Memory Slot number
    logical :: read                       ! ReadOnly status
    logical :: getvm                      ! Memory / File indicator
  end type location
  ! /XCPAR/
```

```

type :: strings
  sequence
    character(len=12) :: type           ! Image Type (see ITYP)
    character(len=12) :: unit           ! Data Units (see IUNI)
    character(len=12) :: code(4)        ! 59 Axis codes (see ICOD)
    character(len=12) :: syst           ! 71 Coordinate system (see ISYS)
    character(len=12) :: name           ! 75 Source name (see ISOU)
    character(len=12) :: line           ! Line name (see ILIN)
end type strings
! /XPAR/
type :: gildas_header
  sequence
    integer(kind=4) :: ityp(3)          ! 1 Image Type
    integer(kind=4) :: form              ! 4 Data format (FMT_xx)
    integer(kind=4) :: nvb               ! 5 Number of blocks
    integer(kind=4) :: fill(5)          ! 6 Reserved space
    integer(kind=4) :: gene              ! 11 General section length
    integer(kind=4) :: ndim              ! 12 Number of dimensions
    integer(kind=4) :: dim(4)            ! 13 Dimensions
    real(kind=8) :: ref1                 ! 17 Reference pixel
    real(kind=8) :: val1                 ! 19 Value at reference pixel
    real(kind=8) :: inc1                 ! 21 Increment per pixel
    real(kind=8) :: ref2                 ! 23 idem for axis 2
    real(kind=8) :: val2                 ! 25
    real(kind=8) :: inc2                 ! 27
    real(kind=8) :: ref3                 ! 29 idem for axis 3
    real(kind=8) :: val3                 ! 31
    real(kind=8) :: inc3                 ! 33
    real(kind=8) :: ref4                 ! 35 idem for axis 4
    real(kind=8) :: val4                 ! 37
    real(kind=8) :: inc4                 ! 39
    !
    integer(kind=4) :: blan              ! 41 Blanking section length
    real(kind=4) :: bval                 ! 42 Blanking value
    real(kind=4) :: eval                 ! 43 Tolerance
    !
    integer(kind=4) :: extr               ! 44 Extrema section length
    real(kind=4) :: rmin                 ! 45 Minimum
    real(kind=4) :: rmax                 ! 46 Maximum
    integer(kind=4) :: min1              ! 47 Pixel of min on axis 1
    integer(kind=4) :: max1              ! 48 Pixel of max .....
    integer(kind=4) :: min2              ! 49 idem for axis 2
    integer(kind=4) :: max2              ! 50
    integer(kind=4) :: min3              ! 51 axis 3
    integer(kind=4) :: max3              ! 52
    integer(kind=4) :: min4              ! 53 axis 4
    integer(kind=4) :: max4              ! 54

```

```

!
integer(kind=4) :: desc           ! 55 Description section length
integer(kind=4) :: iuni(3)        ! 56 Data Unit
integer(kind=4) :: icod(3,4)      ! 59 Axis names
integer(kind=4) :: isys(3)        ! 71 Coordinate System
integer(kind=4) :: dum1           ! Void for padding
!
integer(kind=4) :: posi           ! 74 Position section length
integer(kind=4) :: isou(3)        ! 75 Source name
real(kind=8) :: ra                ! 78 Right Ascension
real(kind=8) :: dec               ! 80 Declination
real(kind=8) :: lii              ! 82 Galactic longitude
real(kind=8) :: bii              ! 84 latitude
real(kind=4) :: epoc             ! 86 Epoch of coordinates
integer(kind=4) :: dum2           ! Void for padding
!
integer(kind=4) :: proj           ! 87 Projection section length
integer(kind=4) :: ptyp           ! 88 Projection type
real(kind=8) :: a0               ! 89 X of projection center
real(kind=8) :: d0               ! 91 Y of projection center
real(kind=8) :: pang             ! 93 Projection angle
integer(kind=4) :: xaxi           ! 95 X axis
integer(kind=4) :: yaxi           ! 96 Y axis
!
integer(kind=4) :: spec           ! 97 Spectroscopy section length
integer(kind=4) :: ilin(3)        ! 98 Line name
real(kind=8) :: fres             !101 Frequency resolution
real(kind=8) :: fima             !103 Image frequency
real(kind=8) :: freq             !105 Rest Frequency
real(kind=4) :: vres             !107 Velocity resolution
real(kind=4) :: voff             !108 Velocity offset
integer(kind=4) :: faxi          !109 Frequency axis
!
integer(kind=4) :: reso           !110 Resolution section length
real(kind=4) :: majo             !111 Major axis
real(kind=4) :: mino            !112 Minor axis
real(kind=4) :: posa            !113 Position angle
!
integer(kind=4) :: sigm           ! 114 Noise section length
real(kind=4) :: noise            ! 115 Theoretical noise
real(kind=4) :: rms              ! 116 Actual noise
!
integer(kind=4) :: prop           ! 117 Proper motion section length
real(kind=4) :: mura             ! 118 along RA, in mas/yr
real(kind=4) :: mudec            ! 119 along Dec, in mas/yr
real(kind=4) :: parallax         ! 120 in mas
!! real(kind=4) :: pepoch         ! 121 in yrs ?

```

```

end type gildas_header
!
type :: sic_header
  sequence
  integer(kind=4) :: ityp(3)           ! 1
  integer(kind=4) :: form              ! 4
  integer(kind=4) :: nvb              ! 5
  integer(kind=4) :: fill(5)          ! 6
  integer(kind=4) :: gene             ! 11
  integer(kind=4) :: ndim             ! 12
  integer(kind=4) :: dim(4)           ! 13
  real(kind=8) :: convert(3,4)        ! 17-40
  !
  integer(kind=4) :: blan             ! 41
  real(kind=4) :: blank(2)           ! 42-43
  !
  integer(kind=4) :: extrema          ! 44
  real(kind=4) :: min                 ! 45
  real(kind=4) :: max                 ! 46
  integer(kind=4) :: where(2,4)       ! 47-54
  !
  integer(kind=4) :: desc              ! 55
  integer(kind=4) :: iuni(3)          ! 56 Unit
  integer(kind=4) :: icod(3,4)        ! 59 Unit1,2,3,4
  integer(kind=4) :: isys(3)          ! 71 System
  integer(kind=4) :: dum1             ! Void
  !
  integer(kind=4) :: posi             ! 74
  integer(kind=4) :: isou(3)          ! 75 Source
  real(kind=8) :: ra                  ! 78
  real(kind=8) :: dec                 ! 80
  real(kind=8) :: lii                 ! 82
  real(kind=8) :: bii                 ! 84
  real(kind=4) :: epoc                ! 86
  integer(kind=4) :: dum2             ! Void
  !
  integer(kind=4) :: proj             ! 87
  integer(kind=4) :: ptyp             ! 88
  real(kind=8) :: a0                  ! 89
  real(kind=8) :: d0                  ! 91
  real(kind=8) :: angle               ! 93
  integer(kind=4) :: x_axis           ! 95
  integer(kind=4) :: y_axis           ! 96
  !
  integer(kind=4) :: spec             ! 97
  integer(kind=4) :: ilin(3)          ! 98 Line
  real(kind=8) :: freqres             ! 101

```

```

real(kind=8) :: freqoff           ! 103
real(kind=8) :: restfre           ! 105
real(kind=4) :: velres            ! 107
real(kind=4) :: veloff           ! 108
integer(kind=4) :: f_axis         ! 109
!
integer(kind=4) :: beam           ! 110
real(kind=4) :: major             ! 111
real(kind=4) :: minor            ! 112
real(kind=4) :: pa               ! 113
!
integer(kind=4) :: sigma          ! 114
real(kind=4) :: noise            ! 115
real(kind=4) :: rms              ! 116
!
integer(kind=4) :: proper         ! 117 Proper motion section length
real(kind=4) :: mura              ! 118 along RA, in mas/yr
real(kind=4) :: mudec            ! 119 along Dec, in mas/yr
real(kind=4) :: parallax         ! 120 in mas
!! real(kind=4) :: pepoch         ! 121 in yrs ?
end type sic_header
!
! Gildas X,Y,Z version
type :: gildas
  sequence
  character(len=256) :: file      ! File name
  type (strings) :: char
  type (location) :: loca
  type (gildas_header) :: gil
  integer(kind=4) :: blc(4)       ! Bottom left corner
  integer(kind=4) :: trc(4)       ! Top right corner
  integer(kind=4) :: header       ! Defined / Undefined
  integer(kind=4) :: status       ! Last error code
  real,           pointer :: r1d(:) ! Pointer to 1D data
  real(kind=8), pointer :: d1d(:)
  integer,        pointer :: i1d(:)
  real,           pointer :: r2d(:, :) ! Pointer to 2D data
  real(kind=8), pointer :: d2d(:, :)
  integer,        pointer :: i2d(:, :)
  real,           pointer :: r3d(:, :, :) ! to 3D data
  real(kind=8), pointer :: d3d(:, :, :)
  integer,        pointer :: i3d(:, :, :)
  real,           pointer :: r4d(:, :, :, :) ! to 4D data
  real(kind=8), pointer :: d4d(:, :, :, :)
  integer,        pointer :: i4d(:, :, :, :)
end type gildas
!

```



```

! SIC Variable version
type :: sic
  sequence
  character(len=256) :: file           ! File name
  type (strings) :: char
  type (location) :: loca
  type (sic_header) :: sic
  integer(kind=4) :: blc(4)
  integer(kind=4) :: trc(4)
  integer(kind=4) :: header
  integer(kind=4) :: status
  real,           pointer :: r1d(:)
  real(kind=8), pointer :: d1d(:)
  integer,        pointer :: i1d(:)
  real,           pointer :: r2d(:, :)
  real(kind=8), pointer :: d2d(:, :)
  integer,        pointer :: i2d(:, :)
  real,           pointer :: r3d(:, :, :)
  real(kind=8), pointer :: d3d(:, :, :)
  integer,        pointer :: i3d(:, :, :)
  real,           pointer :: r4d(:, :, :, :)
  real(kind=8), pointer :: d4d(:, :, :, :)
  integer,        pointer :: i4d(:, :, :, :)
end type sic
!
end module image_def
!

```

Types GILDAS and SIC can be used indifferently to represent the same object. Type GILDAS mimics the old Fortran-77 commons which were used to handle image headers, while type SIC mimics the SIC header variables.

Access to images is very simple. It requires only 3 steps: *i*) to read the header from an existing file, or to create a new header, *ii*) to allocate the data, *iii*) to read or write the data. An example is given below.

```

program image_example
  use image_def                     ! 1
  logical error
  integer ier
  character*32 name1,name2
  !
  type (gildas) :: input_image      ! 2
  real, allocatable :: dinput(:, :), doutput(:, :, :) ! 3
  !
  call gildas_open
  call gildas_char('INPUT$', name1)
  call gildas_char('OUTPUT$', name2)
  call gildas_close

```

```

!
call gildas_null(input_image) ! 4
call sic_parsef (name1,input_image%file,' ','gdf') ! 5
call gdf_read_header (input_image,error) ! 6
if (error) then
    call gagout('E-IMAGE_EXAMPLE, Error opening input file')
    call sysexi(fatale)
endif
allocate(dinput(input_image%gil%dim(1),input_image%gil%dim(2), &
    stat=ier) ! 7
if (ier.ne.0) then
    call gagout('e-image_example, error allocating memory')
    call sysexi(fatale)
endif
call gdf_read_data (input_image, dinput, error) ! 8
if (error) then
    call gagout('E-IMAGE_EXAMPLE, error reading input file')
    call sysexi(fatale)
endif
!
! Create an output image
!-----
call gdf_copy_header (input_image, output_image) ! 9
call sic_parsef (name2,output_image%file,' ','gdf') ! 10
output_image%gil%ndim = 3 ! 11
output_image%gil%dim(1) = input_image%gil%dim(1) ! 11
output_image%gil%dim(2) = input_image%gil%dim(2) ! 11
output_image%gil%dim(3) = 4 ! 11
allocate(doutput(output_image%gil%dim(1), &
    output_image%gil%dim(2),output_image%gil%dim(3), &
    stat=ier) ! 12
if (ier.ne.0) then
    call gagout('E-IMAGE_EXAMPLE, Error allocating memory')
    call sysexi(fatale)
endif
!
! Do something with the data
doutput(:, :, 3) = dinput
!
! Write the output image
call gdf_write_image(output_image,doutput,error) ! 13
if (error) then
    call gagout('E-IMAGE_EXAMPLE, Error writing output file')
    call sysexi(fatale)
endif
!
deallocate(dinput,doutput)

```

end

1. USE the module containing the GILDAS derived type definitions
2. Define the input and output image headers
3. Define the input and output data as allocatable arrays
4. Reset the input image header to “default”
5. Prepare the input file name, `INPUT_IMAGE%FILE`
6. Read the header to initialize the `INPUT_IMAGE` structure.
7. Allocate the data. Note that it is assumed here to be a 2-D array.
8. Read the data, using the information provided in the header structure (in particular the file name).
9. Define the output header, here by making a copy of the input header
10. Setup the output file name
11. Change the output header parameters as needed
12. Allocate the output image data
13. Create and write the output image

The subroutines using GILDAS headers are:

- `SUBROUTINE GILDAS_NULL(HEADER)`  
Reset a header to default values
- `SUBROUTINE GDF_COMPARE_SHAPE(FIRST,SECOND,EQUAL)`  
Compare the shape of the two images defined by the two headers
- `SUBROUTINE GDF_READ_HEADER(IMAG,ERROR)`  
Read the image header. `IMAG%FILE` must have been initialized before.
- `SUBROUTINE GDF_UPDATE_HEADER(IMAG,ERROR)`  
Update an image header: write the modified header to the image file.
- `SUBROUTINE GDF_COPY_HEADER(INPUT,OUTPUT)`  
Copy an input header structure to an output header structure
- `SUBROUTINE GDF_TRANSPOSE_HEADER(INPUT,OUTPUT,ORDER,ERROR)`  
Transpose a header according to given transposition code. This routine transposes the axes information, but not the data.

### 3.3 Obsolescent Fortran-77 access routines

In Fortran-77, neither virtual memory handling, nor data structures, are part of the language. We had written a number of routines to access images. These routines are more complex than the new Fortran-90 method of access, and are provided here just for completeness.

To read a file requires six basic steps:

- allocation of an *Image Slot* using `GDF_GEIS`.
- connection of an image to the slot using `GDF_REIS` or `GDF_WRIS`.
- reading the header using `GDF_RHSEC`
- connection of a *Memory Slot* to the image (or a subset of it) using `GDF_GEMS`
- once the *Memory Slot* is no longer needed, disconnection using `GDF_FRMS`
- once the image slot is no longer required, disconnection using `GDF_FRIS`

Creation of a new file is slightly different:

- allocation of an *Image Slot* using `GDF_GEIS`
- preparatin of the header using e.g. `GDF_WHSEC`
- creation of the image to the slot using `GDF_CRIS`. The header is written at this stage.
- connection to the image (or a subset of it) using `GDF_GEMS`
- once the *Memory Slot* is no longer needed, disconnection using `GDF_FRMS`. This is the step which actually does the writing.
- once the image slot is no longer required, disconnection using `GDF_FRIS`

#### 3.3.1 Image Slot Handling

SUBROUTINE `GDF_GEIS` (`IS`,`ERROR`)

GDF	GEt Image Slot			
IS	I	Slot number		Output
ERROR	L	Error flag		Output

`GDF_GEIS` returns a unique image slot number `IS`, for further use in all other `GDF_xxxx` routines.

SUBROUTINE `GDF_CLIS` (`IS`,`ERROR`)

GDF	CLOse Image Slot			
IS	I	Slot number		Input
ERROR	L	Error flag		Output

`GDF_CLIS` close an image slot `IS`, disconnecting all associated memory slot after flushing all pending updates on the image. The image slot `IS` stays reserved, but available for further use.

SUBROUTINE GDF\_FRIS (IS,ERROR)

GDF FRee Image Slot			
IS	I	Slot number	Input
ERROR	L	Error flag	Output

GDF\_CLIS free an image slot IS, disconnecting all associated memory slot after flushing all pending updates on the image. The image slot IS should no longer be referenced in any call after this operation.

INTEGER FUNCTION GDF\_STIS(IS)

GDF SStatus of Image Slot			
IS	I	Slot number	Input

Return status of slot: -1 No such slot, 0 Empty (Not allocated), 1 Read (Opened for ReadOnly), 2 Full (Allocated but not opened), 3 Write (Opened for ReadWrite).

### 3.3.2 Image Connection

Once a slot is reserved, it must be associated with an image. The following 4 routines connect an image to a specified image slot IS, in one of the 4 possible access modes (CReate, EXtend, REad only, Write and Read). No part of the image is immediately accessible after that, but the image slot is ready for memory connection using GDF\_GEMS.

SUBROUTINE GDF\_CRIS (IS,GTYPE,NAME,FORM,SIZE,ERROR)

GDF CReate Image Slot			
Arguments:			
IS	I	Slot number	Input
GTYPE	C*(*)	Image type	Input
NAME	C*(*)	File name	Input
FORM	I	Image format (Real,Integer...)	Input
SIZE	I	Image size	Input
ERROR	L	Error flag	Output

Creates a new file from the information available in image slot IS. The internal header should have been defined (through calls to GDF\_WHSEC) before, and is written on the new file header.

SUBROUTINE GDF\_EXIS (IS,GTYPE,NAME,FORM,SIZE,ERROR)

GDF EXtend Image Slot			
Arguments:			
IS	I	Slot number	Input
GTYPE	C*(*)	Image type	Input
NAME	C*(*)	File name	Input
FORM	I	Image format (Real,Integer...)	Input
SIZE	I	Image size	Input
ERROR	L	Error flag	Output

This routine is used to change the size of an image, usually by increasing the last dimension of the image. The new internal header must have been defined before in image slot IS, by use of the GDF\_WHSEC routine.

SUBROUTINE GDF\_REIS (IS,GTYPE,NAME,FORM,SIZE,ERROR)

GDF REad Image Slot

Arguments:

IS	I	Slot number	Input
GTYPE	C*(*)	Image type	Output
NAME	C*(*)	File name	Input
FORM	I	Image format (Real,Integer...)	Output
SIZE	I	Image size	Output
ERROR	L	Error flag	Output

This routine is used to open an existing image for ReadOnly operations. The image is available for file sharing.

SUBROUTINE GDF\_WRIS (IS,GTYPE,NAME,FORM,SIZE,ERROR)

GDF WRite Image Slot

Arguments:

IS	I	Slot number	Input
GTYPE	C*(*)	Image type	Output
NAME	C*(*)	File name	Input
FORM	I	Image format (Real,Integer...)	Output
SIZE	I	Image size	Output
ERROR	L	Error flag	Output

This routine is used to open an existing image for ReadWrite operations. The image is not available for file sharing.

### 3.3.3 Memory Connection

After an image is connected, a part of it should be brought into virtual memory. The access type for this virtual memory area (ReadOnly or ReadWrite) depends on the type of image connection.

SUBROUTINE GDF\_GEMS (MS, IS, BLC, TRC, ADDR, FORM, ERROR)

GDF Get Memory Slot

MS	I	Memory Slot number	Output
IS	I	Image Slot number	Input
BLC	I(4)	Bottom Left Corner	Input
TRC	I(4)	Top Right Corner	Input
ADDR	I	Virtual memory address	Output
FORM	I	Type of data	Input
ERROR	L	Logical error flag	Output

GDF\_GEMS reads a subset of the image connected to slot IS into a memory area (memory slot MS) located at address ADDR. The image subset is "incarnated" into the specified data type FORM, which may be different from the image data type. When BLC(i) and TRC(i) are set to zero, they default to the current image dimension (i.e BLC(i)=1 and TRC(i)=Dim(i)). After this routine, virtual memory address ADDR is the start of an array of type FORM containing an incarnation of the image subset that can be used for further processing.

```
SUBROUTINE GDF_FRMS (MS,ERROR)
```

```
-----
GDF  FRee Memory Slot
-----
```

GDF\_FRMS Frees the memory slot MS. If the connected image was connected with write access, the image is updated. The memory slot is no longer available after this operation.

```
SUBROUTINE GDF_UPMS (MS,ERROR)
```

```
-----
GDF  UPDate Memory Slot
-----
```

GDF\_UPMS Updates the image connected to the memory slot MS. The memory slot is left unmodified. This routine is provided for safety measures.

The I/O (or mapping) are thus only done in the GDF\_xxMS routines, in a system dependent way. All data format translation (incarnation in a different type, or machine dependent data types) are done at this level. This allows a transparent operation in an heterogeneous environment.

### 3.3.4 Image Header Access

Each image slot has an associated buffer area, from which header information can be retrieved. This buffer, named HEADER(\*,IS) for slot IS, is an array of size is 128 words, containing an copy of the file header, but converted to the machine-dependent number format. Subroutines are available to read (or write) the header section from (to) these internal buffers: GDF\_RHSEC and GDF\_WHSEC.

To provide pre-defined data structures with the appropriate header item type, 3 named commons called X, Y or Z are described in include files `inc:apar.inc`, where "a" is the common "name". Routines are available to copy headers from one common to another (GDF\_CHab), and to read header from (or copy header to) the internal array HEADER (GDF\_READa, GDF\_WRITa).

Each named common contains a variable to define one memory slot, `a_MSL0`, but more memory slots can be used simultaneously by declaring other variables in the application program. Two routines can be used to communicate between the "new" Fortran-90 header structures and the "old" Fortran-77 named commons:

```
SUBROUTINE GDF_HEADER_TO_COMMON (IMAG,COM)
SUBROUTINE GDF_COMMON_TO_HEADER (IMAG,COM)
```

where IMAGE is a header structure, and COM is 'X', 'Y' or 'Z'

**Named Common Description** The named commons have the following structure, given here for X common. Replace X by Y or Z respectively for the others. Note that each named common is in fact made of 3 different FORTRAN commons, to handle different types of information. Two representations of the character strings exist, one in XCPAR as character variables, one in XPAR as integer arrays. The application program should deal only with the XCPAR representation.

```

INTEGER MXCPAR,MXIPAR,MXPAR
PARAMETER (MXPAR=128)      ! 128 header integers
PARAMETER (MXIPAR=5)       ! 5 Extra integers
PARAMETER (MXCPAR=9)       ! 9 character*12 strings
CHARACTER*120 X_FILE       ! File name
CHARACTER*12 X_TYPE        ! Image type
CHARACTER*12 X_UNIT        ! Image unit
CHARACTER*12 X_CODE(4)     !      Axis type
CHARACTER*12 X_SYST        ! System type
CHARACTER*12 X_NAME        ! Source name
CHARACTER*12 X_LINE        ! Line name
COMMON /XCPAR/ X_FILE,
+      X_TYPE,X_UNIT,X_CODE,X_SYST,X_NAME,X_LINE
INTEGER X_SIZE             ! image size
INTEGER X_ADDR             ! start map address
INTEGER X_ISLO             ! Image slot
INTEGER X_MSLO             ! Memory Slot
LOGICAL X_READ             ! Read Only status
COMMON /XIPAR/ X_SIZE,X_ADDR,X_ISLO,X_MSLO,X_READ
INTEGER X_BUFF(MXPAR)
REAL*8 X_CONV(3,4)
EQUIVALENCE (X_CONV(1,1),X_REF1)
EQUIVALENCE (X_ITYP,X_BUFF)

*
* /XPAR/
      INTEGER X_ityp(3)    ! Image Type TYPE
      INTEGER*4 X_form     ! Image Format FORM
      INTEGER*4 X_nvb      ! Number of blocks of image
      INTEGER*4 X_fill(5)  ! Reserved space

*
      INTEGER X_GENE       ! Size of GENERAL section
      INTEGER X_NDIM       ! Number of dimensions
      INTEGER X_DIM(4)     ! Dimensions
      REAL*8 X_REF1        ! Reference Pixel for First Axis
      REAL*8 X_VAL1        ! Value at Reference Pixel
      REAL*8 X_INC1        ! Increment per Pixel
      REAL*8 X_REF2        ! As above for Second Axis
      REAL*8 X_VAL2
      REAL*8 X_INC2
      REAL*8 X_REF3
      REAL*8 X_VAL3
      REAL*8 X_INC3
      REAL*8 X_REF4
      REAL*8 X_VAL4
      REAL*8 X_INC4

*
      INTEGER X_BLAN       ! Length of Blanking Section

```



```

REAL X_BVAL          ! Blanking Value
REAL X_EVAL          ! Tolerance on Blanking
*
INTEGER X_EXTR        ! Length of Extrema Section
REAL X_RMIN           ! Minimum
REAL X_RMAX           ! Maximum
INTEGER X_MIN1        ! Pixel of Minimum (first axis)
INTEGER X_MAX1        ! " of Maximum
INTEGER X_MIN2        ! " " Minimum (second axis)
INTEGER X_MAX2
INTEGER X_MIN3
INTEGER X_MAX3
INTEGER X_MIN4
INTEGER X_MAX4
*
INTEGER X_DESC        ! Description section
INTEGER X_IUNI(3)     ! Unit Name X_UNIT
INTEGER X_ICOD(3,4)   ! Axis Name X_CODE(4)
INTEGER X_ISYS(3)     ! Coordinate System X_SYST
*
INTEGER X_POSI        ! Position section
INTEGER X_ISO(3)      ! Source Name X_SOUR
REAL*8 X_RA           ! Right Ascension (Radians)
REAL*8 X_DEC          ! Declination (Radians)
REAL*8 X_LII          ! Galactic Longitude (Radians)
REAL*8 X_BII          ! Galactic Latitude (Radians)
REAL*4 X_EPOCH        ! Epoch of RA and DEC (Years)
*
INTEGER X_PROJ        ! Projection Section
INTEGER X_PTYPE       ! Projection Type
REAL*8 X_A0           ! Longitude of Projection Center
REAL*8 X_D0           ! Latitude of Center (Radians)
REAL*8 X_PANG         ! Position Angle
INTEGER X_XAXI        ! First Projected axis
INTEGER X_YAXI        ! Second Projected axis
*
INTEGER X_SPEC        ! Spectroscopy Section
INTEGER X_ILIN(3)     ! Line name X_LINE
REAL*8 X_FRES         ! Frequency Resolution (MHz)
REAL*8 X_FIMA         ! Image Frequency (MHz)
REAL*8 X_FREQ         ! Rest Frequency (MHz)
REAL*4 X_VRES         ! Velocity resolution (km/s)
REAL*4 X_VOFF         ! Velocity Offset (km/s)
INTEGER X_FAXI        ! Frequency axis (MHz)
*
INTEGER X_RESO        ! Resolution Section
REAL*4 X_MAJO        ! Major Axis (User Units)

```

```

REAL*4 X_MINO      ! Minor Axis (User Units)
REAL*4 X_POSA      ! Position Angle (Radians)
COMMON /XPAR/X_ITYP,X_FORM,X_NVb,X_FILL,X_GENE,X_NDIM,X_DIM,
+   X_REF1,X_VAL1,X_INC1,X_REF2,X_VAL2,X_INC2,X_REF3,X_VAL3,
+   X_INC3,X_REF4,X_VAL4,X_INC4,X_BLAN,X_BVAL,X_EVAL,X_EXTR,
+   X_RMIN,X_RMAX,X_MIN1,X_MAX1,X_MIN2,X_MAX2,X_MIN3,X_MAX3,
+   X_MIN4,X_MAX4,X_DESC,X_IUNI,X_ICOD,X_ISYS,X_POSI,X_ISOU,
+   X_RA,X_DEC,X_LII,X_BII,X_EPOC,X_PROJ,X_Ptyp,X_AO,X_DO,
+   X_PANG,X_XAXI,X_YAXI,X_SPEC,X_ILIN,X_FRES,X_FIMA,X_FREQ,
+   X_VRES,X_VOFF,X_FAXI,X_RESO,X_MAJO,X_MINO,X_POSA

```

### Basic Header Handling

```
SUBROUTINE GDF_UPIH (IS,NAME,ERROR)
```

```
-----
GDF  Update Image Header
```

```
      Update an image header
```

```
Arguments :
```

IS	I	Image slot	Input
NAME	C(*)	Gildas Data Frame name	Input
ERROR	L	Logical error flag	Output

```
Subroutines :
```

```
-----
SUBROUTINE GDF_RHSEC(IS,TYPE,ARRAY,SIZE,ERROR)
```

```
-----
GDF  Read Header Section
```

```
      Find information Gildas Data Frame header
```

```
Arguments
```

IS	I	Image slot number	Input
TYPE	C*12	Information wanted	Input
ARRAY	I (*)	Array to receive information	Output
LENGTH	I	Size of array	Input/Output
ERROR	L	Logical error flag	Output

```
Subroutines :
```

```
      SIC_AMBIGS, R4TOR4
-----
```

```
SUBROUTINE GDF_WHSEC(IS,TYPE,ARRAY,SIZE,ERROR)
```

```
-----
GDF  Write Header SEction
```

```
      Write a descriptor in Header of file.
```

```
Arguments :
```

IS	I	Image slot	Input
TYPE	C(*)	Type of section	Input
ARRAY	I(*)	Address of section	Input
LENGTH	I	Length of section (Words)	Input
ERROR	L	Logical error status	Output

Subroutines :

SIC\_AMBIGS, R4TOR4

---

### Named Commons Handling

SUBROUTINE GDF\_READX (Islo,Error)

SUBROUTINE GDF\_READY (Islo,Error)

SUBROUTINE GDF\_READZ (Islo,Error)

---

GDF Read Header of the X, Y or Z common area from image slot Islo

ISLO Integer Image Slot number

ERROR Logical Error Flag

---

SUBROUTINE GDF\_WRITX (Islo,Error)

SUBROUTINE GDF\_WRITY (Islo,Error)

SUBROUTINE GDF\_WRITZ (Islo,Error)

---

GDF Write Header from the X, Y or Z common area to image slot Islo

ISLO Integer Image Slot number

ERROR Logical Error Flag

---

SUBROUTINE GDF\_CHXY, GDF\_CHYZ, GDF\_CHXZ

---

GDF Copy Header of the X, Y or X common into Y, Z, Z respectively  
and move memory pointers accordingly.

---

## 3.4 Creating GILDAS Tasks

GILDAS tasks (e.g. for task “taskname”) are composed of 4 different parts :

- The program, `taskname.exe`.
- A help text file, `taskname.hlp`
- The initialization file, `taskname.init`, read by commands `RUN` and `SUBMIT` to get from the user the task input parameters.
- Optionally, a checker file, `taskname.check`, read by commands `RUN` and `SUBMIT`, to check the input parameters and pass them to the task. It is optional: SIC will use a suitable default derived from the initialisation file when the checker file does not exist.

Tasks and their associated files are searched for in the `task#dir:` search path (by default “./;gag.tasks:”).

The structure of the program should always be the same :

- An input parameters definition part. No image creation or access should be made before all parameters are defined.
- An input and output image mapping part. If possible all necessary images should be defined in this part.
- One or more subroutines performing the processing. No new image should be created or accessed in the subroutines. The subroutines should preferably not make any reference to any common.
- A cleanup part. An error status should be returned to the system in case of error through a call to SYSEXI.

### 3.5 A Template Task

The following program is a typical task with two input images and an output one. The easiest way to create a new task is just to start with this working template and modify what is needed.

#### 3.5.1 Source code

```

program combine
!-----
! GILDAS Combine in different ways two input images
!      (or data cubes)...
!-----
use image_def
use gbl_format
!
character*80 namex,namey,namez, code*20
logical error
real ay,az,ty,tz,b,c
type (gildas) :: hx, hy, hz
real, allocatable :: dx(:,,:), dy(:,,:), dz(:,)
integer i, j, n, m, ier
!
call gildas_open
call gildas_char('Z_NAME$',namez)
call gildas_real('Z_FACTOR$',az,1)
call gildas_real('Z_MIN$',tz,1)
call gildas_char('Y_NAME$',namey)
call gildas_real('Y_FACTOR$',ay,1)
call gildas_real('Y_MIN$',ty,1)
call gildas_char('X_NAME$',namex)
call gildas_real('BLANKING$',b,1)
call gildas_real('OFFSET$',c,1)
call gildas_char('FUNCTION$',code)
call gildas_close
!
n = len_trim(namez)

```

```

if (n.eq.0) goto 100
call gildas_null(hz)
call sic_parsef(namez(1:n),hz%file,' ','gdf')
call gdf_read_header(hz,error)
if (error) then
  call gagout('F-COMBINE, Cannot read input file')
  goto 100
endif
!
n = len_trim(namey)
if (n.eq.0) goto 100
call gildas_null(hy)
call sic_parsef(namey(1:n),hy%file,' ','gdf')
call gdf_read_header(hy,error)
if (error) then
  call gagout('F-COMBINE, Cannot read input file')
  goto 100
endif
!
if (hz%gil%eval.ge.0.0) hz%gil%eval = &
  max(hz%gil%eval,abs(hz%gil%bval*1e-7))
if (hy%gil%eval.ge.0.0) hy%gil%eval = &
  max(hy%gil%eval,abs(hy%gil%bval*1e-7))
!
! Check input dimensions
do i=1,4
  if (hy%gil%dim(i).ne.hz%gil%dim(i)) then
    n = 1
    do j=i,4
      n = n*hz%gil%dim(j)
    enddo
    if (n.ne.1) then
      call gagout('F-COMBINE, Input images are non coincident')
      goto 100
    else
      call gagout('W-COMBINE, Combining a cube with a plane')
    endif
  endif
enddo
!
call gdf_copy_header(hy,hx)
n = len_trim(namex)
if (n.eq.0) goto 100
call sic_parsef(namex(1:n),hx%file,' ','gdf')
hx%gil%blan = 2
hx%gil%bval = b
hx%gil%eval = 0.0

```

```

hx%gil%extr = 0
!
! Allocate the arrays. Note that the allocated arrays do not conform
! to the shape of the images: DZ is allocated as a 1-D array, DX,DY
! as 2-D arrays, possibly of second dimension 1.
!
n = hz%loca%size
m = hx%loca%size/hz%loca%size
allocate(dx(n,m),dy(n,m),dz(n),stat=ier)
if (ier.ne.0) then
    call gagout('F-COMBINE, Input images are non coincident')
    goto 100
endif
!
! Read the input data
call gdf_read_data(hz,dz,error)
call gdf_read_data(hy,dy,error)
!
if (code.eq.'ADD') then
    call add002(dz,dy,dx, &
        n,m, &
        hz%gil%bval,hz%gil%eval,az,tz, &
        hy%gil%bval,hy%gil%eval,ay,ty, &
        hx%gil%bval,c)
elseif (code.eq.'DIVIDE') then
    call div002(dz,dy,dx, &
        n,m, &
        hz%gil%bval,hz%gil%eval,az,tz, &
        hy%gil%bval,hy%gil%eval,ay,ty, &
        hx%gil%bval,c)
elseif (code.eq.'MULTIPLY') then
    call mul002(dz,dy,dx, &
        n,m, &
        hz%gil%bval,hz%gil%eval,az,tz, &
        hy%gil%bval,hy%gil%eval,ay,ty, &
        hx%gil%bval,c)
elseif (code.eq.'OPTICAL_DEPTH') then
    call opt002(dz,dy,dx, &
        n,m, &
        hz%gil%bval,hz%gil%eval,az,tz, &
        hy%gil%bval,hy%gil%eval,ay,ty, &
        hx%gil%bval,c)
else
    call gagout('Invalid operation code '//code)
    goto 100
endif
!

```

```

! Write output file
call gdf_write_image(hx,dx,error)
!
stop 'S-COMBINE, Successful completion'
!
100      call sysexi (fatale)
end
!
subroutine add002(z,y,x,n,m,bz,ez,az,tz,by,ey,ay,ty,bx,c)
!-----
! GDF Internal routine
!   Linear combination of input arrays
!    $X = Ay*Y + Az*Z + C$ 
! Arguments
!   Z          R*4(*)  Input array (N)
!   Y          R*4(*)  Input array (N,M)
!   X          R*4(*)  Output array (N,M)
!   N,M        I       Dimensions of arrays
!   BX,BY,BZ   R*4     Blanking values
!   EY,EZ      R*4     Tolerance on blanking
!   AZ,AY      R*4     Multiplicative factor of array Z, Y
!   TZ,TY      R*4     Threshold on Z,Y
!   C          R*4     Additive constant
!-----
integer :: n                !
real :: z(n)               !
integer :: m                !
real :: y(n,m)             !
real :: x(n,m)             !
real :: bz                 !
real :: ez                 !
real :: az                 !
real :: tz                 !
real :: by                 !
real :: ey                 !
real :: ay                 !
real :: ty                 !
real :: bx                 !
real :: c                  !
! Local
integer :: i,k
!
do k=1,m
  do i=1,n
    if (abs(z(i)-bz).gt.ez .and. abs(y(i,k)-by).gt.ey &
      & .and. z(i).gt.tz .and. y(i,k).gt.ty) then
      x(i,k) = ay*y(i,k) + az*z(i) + c
    end if
  end do
end do

```

```

        else
            x(i,k) = bx
        endif
    enddo
enddo
end

```

### 3.5.2 Initialization file

The initialization file is a standard SIC procedure containing only commands from the languages TASK\ or SIC\. TASK\ language must be specified explicitly. The command syntax is always the following

```
TASK\Type_of_parameter "Prompt string" Parameter$[Dimension]
where
```

- Type\_of\_Parameter can be CHARACTER, REAL, INTEGER or LOGICAL
- "Prompt string" is a text used as a prompt if required
- Parameter\$ is the parameter name, the parameter being a standard SIC variable, possibly with one dimension. It is recommended to include a \$ as last character to avoid possible confusion with user declared variables.

```

!
! Combine.INIT
TASK\FILE "First input map" Z_NAME$
TASK\REAL "Scaling factor" Z_FACTOR$
TASK\REAL "Threshold" Z_MIN$
TASK\FILE "Second input map" Y_NAME$
TASK\REAL "Scaling factor" Y_FACTOR$
TASK\REAL "Threshold" Y_MIN$
TASK\FILE "Output map" X_NAME$
TASK\REAL "New blanking value" BLANKING$
TASK\REAL "Output offset" OFFSET$
TASK\CHARACTER "Function" FUNCTION$
TASK\GO                      ! Must be last command

```

The parameter names, types and dimensions must correspond to those declared in the source code. All parameters must be defined. Ordering may be important (see Checker File).

### 3.5.3 The HELP file

A standard help file should be prepared for each task. The format follows that of SIC help files. Topics are identified by a "1" in first column, and subtopics by a "2". The help file must start with the task name as main topic, and must have a subtopic for each parameter. More subtopics may exist. The help file must be in the same place as the executable image, and have file type .hlp. For the example above, combine.hlp contains

```
1 COMBINE
```

```
COMBINE
```



It makes "combinations" of two input images to produce a third one. The two input images may have the same dimensions, or the first one (Z one) may have less dimensions than the second (Y) one. In the latter case, combinations will occur for all the extra planes of the Y image. For example you can divide all the plane of an input (Y) 3-D cube by a 2-D (Z) image, provided each plane of the cube matches the single image...

Operations are

ADD  $X = Ay*Y + Az*Z + C$

MULTIPLY  $X = Ay*Y * Az*Z + C$

DIVIDE  $X = Ay*Y / Az*Z + C$

provided  $Y > Ty$  and  $Z > Tz$ , where  $Ty$  and  $Tz$  are thresholds set by parameters  $YMIN\$$  and  $ZMIN\$$ .

2 Z\_NAME\$

This is the name of the input map with the smaller number of dimensions.

2 Z\_FACTOR\$

This is a scaling factor for map Z\_NAME\$.

2 Z\_MIN\$

This is a threshold on map Z\_NAME\$.

2 Y\_NAME\$

This is the name of the input map with the larger number of dimensions.

2 Y\_FACTOR\$

This is a scaling factor for map Y\_NAME\$.

2 Y\_MIN\$

This is a threshold on map Y\_NAME\$.

2 X\_NAME\$

This is the name of the output map.

2 BLANKING\$

This is the blanking value chosen for the output map.

2 OFFSET\$

This is an offset added to the output map.

2 FUNCTION\$

Selected operation. Possible operations are ADD, MULTIPLY, DIVIDE (Y by Z).

### 3.5.4 Checker File

The checker file is optional. If it does not exist, SIC will create one writing all the task input variables in the order in which they have been defined.

The checker file is another SIC procedure containing only TASK\ and SIC\ commands, which tests the validity of the input parameters (to avoid submission of tasks with bad parameters), and writes the parameters. Checking is optional and can be done using SIC facilities. All parameters are known SIC variables. Writing is done using command TASK\WRITE, and the ordering must

match the source code. The task is initiated by the TASK\GO command.

```

!
! Combine.CHECK
SIC\IF (FUNCTION$.EQ."ADD") THEN
    SIC\SAY "Computing X_NAME = -
        'Y_FACTOR$'*Y_NAME + 'Z_FACTOR$'*Z_NAME + 'OFFSET$'"
SIC\ELSE IF (FUNCTION$.EQ."DIVIDE") THEN
    SIC\SAY "Computing X_NAME = -
        'Y_FACTOR$'*Y_NAME / 'Z_FACTOR$'*Z_NAME + 'OFFSET$'"
SIC\ELSE IF (FUNCTION$.EQ."MULTIPLY") THEN
    SIC\SAY "Computing X_NAME = -
        'Y_FACTOR$'*Y_NAME * 'Z_FACTOR$'*Z_NAME + 'OFFSET$'"
SIC\ELSE
    SIC\SAY "Invalid operation 'FUNCTION$'"
    SIC\RETURN    ! Return without a GO command : no submission
SIC\ENDIF
!
TASK\WRITE Z_NAME$
TASK\WRITE Z_FACTOR$
TASK\WRITE Z_MIN$
TASK\WRITE Y_NAME$
TASK\WRITE Y_FACTOR$
TASK\WRITE Y_MIN$
TASK\WRITE X_NAME$
TASK\WRITE BLANKING$
TASK\WRITE OFFSET$
TASK\WRITE FUNCTION$
TASK\GO
! Effectively RUNs or SUBMITs the task.

```

### 3.6 Debugging Tasks

In addition to the predefined directory, GILDAS\_RUN:, another directory GILDAS\_LOCAL: is also searched for tasks by commands VECTOR\RUN and VECTOR\SUBMIT. This area is searched before GILDAS\_RUN:. If the required task is found here, the initialisation, checker and help files should also be present in the same directory. This feature allows to have experimental or user-private tasks.

Although the tasks are supposed to be non-interactive programs spawned or submitted from a main program, interactive use is possible for debugging purpose. If activated interactively, a task will ask one question for each parameter, specifying the parameter type, the parameter name, and the parameter dimension. No more information will be available. The initialization and checker file are not needed for this.

Moreover since all algorithms can be standard subroutines due to the use of virtual memory, the GILDAS interface can be done easily after debugging the algorithm.

## Chapter 4

# GTV Programming Manual

### 4.1 Concept

The GTVIRT is a fast low-level Graphic library, allowing to develop Graphic programs in a completely device independent way. The GTVIRT library involves the following concepts:

- *the Plot Page*  
The Plot Page is a virtual workspace on which all graphic items will be (virtually) drawn. The Plot Page units are “virtual centimeters”. Actual drawing on a graphic device is usually done with an automatic scaling factor to match the Plot Page to the device. For hardcopy outputs, an exact matching between the Plot Page unit and centimeters is possible.
- *Segments*  
Graphic segments are the smallest separable entities in a drawing. Segments are named, and can be edited to change their aspects. Although segments are usually associated with a single, easily identified, part of the drawing (e.g. a curve, or a label, or a bitmap image), they can contain anything. The segmentation is not defined by the GTVIRT, but left to the calling program. Segments are named uniquely to help identifying them.
- *Coordinate Systems* Two coordinate systems are available within the GTVIRT: a Page coordinate system (whose units are centimeters), and a User coordinate system. Drawing can be done in any of these systems.
- *Directories*  
Directories are special graphic segments used to group in a logical way ensembles of segments. A complete directory tree can be specified in a drawing. As for the segments, the structure of the drawing is defined by the calling program, not by the GTVIRT library.  
Each directory has its own User coordinate system. This feature can be used to map several coordinate systems to different or similar regions of the Plot Page.
- *Devices*  
The GTVIRT is (completely ?) device independent. A drawing can be prepared and visualized separately. The GTVIRT allows display on a large variety of graphic devices.
- *Metafiles*  
Directories and directory trees of a Plot can be saved on a (binary) metafile, complete with all graphic segments and user coordinate systems. Metafiles can be imported as Sub-directories in any Plot.

- *Hardcopy*  
“Hard” copy on paper-like Graphic devices such as pen plotters, laser printers, etc..., are completely transparent in the GTVIRT. “Soft” hardcopy on files in industry standard graphic languages such as PostScript or HPGL is possible too.
- *Windows*  
*Windows* are available on some devices. This includes X-Window terminals, and MS-Windows screen. *Windows* can be attached to any directory. Several windows can be attached to the same directory. A *Window* displays all the sub-directories depending on the directory to which it is attached.
- *Plotting Depth*  
Each segment can be understood as an opaque or transparent plot. The order in which opaque segments are plotted can be controlled by their plotting depth. This feature is important when plotting bitmaps or drawing colour filled curves.
- *8-bit images*  
are supported on Windows graphic displays (X-Window or MS-Window or MAC).

The GTVIRT normally operates in a buffered mode. All drawings command a written into an internal metacode, and only transmitted to the plotting device when required by the calling program. In addition, for applications which require an unbuffered plot, an *immediate* mode is available.

## 4.2 Programming

- Initialisation  
The GTVIRT library is initialized by a a call to  
SUBROUTINE INIT\_GTVIRT  
followed by a call to SUBROUTINE GTINIT(LX,LY,LUNG,LUNH,NAME,USER\_ROUTINE  
where
  - LX,LY are the page dimensions
  - LUNG,LUNH are logical unit numbers for graphic output and hardcopies, to be supplied by the caller
  - NAME is a character string giving the name of the top directory of the drawings. NAME should start with a < sign.
  - USER\_ROUTINE is a user-supplied subroutine called when moving from a sub-directory to the other. It must provide to the GTVIRT information about coordinate systems. The only suitable routine is available in the GREG program, and is named GREG\_USER
- The GTVIEW subroutine

The GTVIEW subroutine is one of the main control routine in the library. It is used to send drawing commands from the internal metacode to the graphic device, but also to performs other actions on the metacode.

SUBROUTINE GTVIRT(Mode)

where Action is a character string which can be

- 'Append'  
Plot all metacode starting from last drawn vector
- 'Rewind'  
Clear the screen, and plot the whole metacode from first vector
- 'Update'  
Update the screen (all windows) if needed
- 'Limits'  
Recompute the plot limits (Bounding Box)
- 'Purge'  
Delete images associated to all windows
- 'Delete'  
Delete images associated to the current window
- 'Zap'  
Delete cache-bitmap associated to images
- 'Sleep'  
Set screen update off
- 'Wake'  
Set screen update on

### 4.3 Basic Sequence

GREG and the GTVIRT are intimately related, and it is not recommended to use the GTVIRT without GREG. Accordingly, the programming example given below also uses some (primitive) GREG subroutines. The basic drawing sequence is

```
CALL GR_SEGM (Segment_Name,Error)    ! 1
CALL GTPOLYL (N,Xarray,Yarray)      ! 2
CALL ...                             ! 3
CALL GTVIEW('Append')                ! 4
```

1. Open a new graphic segment. This routine is an “intelligent” routine checking whether pen attributes have been changed, and calling both **GTSEGM** (the basic segment creation routine) and **GTEDIT** (which defines the segment attributes like pen colour, dashed pattern, thickness).
2. Use any drawing routine you wish, e.g. an polyline. The drawing commands go to the internal metacode only at that time.
3. ...
4. Update the graphic screen with all the new drawing commands which have been put in the metacode since last call to **GTVIEW('Append')**

Further buffering between graphic segments can be obtained by enclosing a set of complete sequences like the above one between a call to **GTVIEW('Sleep')** and **GTVIEW('Wake\_up')**. Drawing to the screen will then only happen when the call to **GTVIEW('Wake\_up')** is made. For compatibility with other subroutines which may also perform their own **Sleep / Wake\_up** control, it is recommended to use the logical function **GTSTAT** instead, e.g.

```

SLEEP = GTSTAT ('Sleep')

... calls to GTVIRT ...

IF (.NOT.SLEEP) CALL GTVIEW('Wake_up')

```

## 4.4 Plot Structuration and multi-window applications

For some complex applications, it is useful to be able to create several graphic windows and display different plots in each of them. Such a control over the plot structure is preferably done using the GTVL command language, by appropriate calls to subroutine GR\_EXECL.

SUBROUTINE GR\_EXECL(Command)

execute a command from the GTVL command language.

The most relevant commands to structure a plot are

- CREATE DIRECTORY <NAME to create new top directory called <NAME
- CHANGE DIRECTORY <NAME to move to this (top) directory
- CREATE WINDOW to create a new window associated to the current directory
- CHANGE POSITION Code to move the current window a given position in the screen.
- CLEAR WINDOW to clear a window
- CLEAR TREE to erase the tree linked to the current directory (from the current top directory).
- CREATE DIRECTORY NAME to create new sub directory
- CHANGE DIRECTORY NAME to move to this sub directory

Refer to the GREG manuals for detailed help on these commands.

In addition, subroutine GTEXIST is useful to check whether a given directory already exists.

## 4.5 Subroutines

- gtalph  
Commute to alphanumeric mode. On Windows system, transfer focus to the terminal window.
- gtchar  
Draws a character string.
- gtclal  
Clear alphanumeric screen. On Windows applications, transfer focus to the (current) graphic window, and raise in on top of others to display it.
- gtclear  
Erase the whole plot, all directory structures, destroy all associated windows, etc... An empty top directory is then re-created
- gtclos  
Close the current graphic device

- `gtclpl`  
Raise alphanumeric window, and returns focus to it.
- `gtcurs`  
Call the cursor
- `gtdls`  
Delete the last graphic segment
- `gtdraw`  
Draw a vector from current pen position to the specified point.
- `gtdedit`  
Edit the current segment properties.
- `gterflag`, `gterrsto`, `gterrst`  
Control error status of the library
- `gtexist`  
Controls the existence of a named sub-directory.
- `gthard`  
Create a hardcopy
- `gtinit`  
Initialize the GTVIRT drawing space and the GTVL language
- `gtopen`  
Open a graphic device
- `gtpolyl`  
Draw a set of lines.
- `gtreloc`  
Move current pen position to specified coordinates.
- `gtsegm`  
Open a new segment
- `gtview`  
Activate the drawing
- `gtwhere`  
Returns current pen position
- `gtg_charlen`  
Computes character string length
- `gtg_charsiz`  
Returns character size
- `gtg_curs`  
Returns cursor existence

- `gtg_screen`  
Returns clipping parameters
- `gtg_open`  
Returns device parameters
- `gtstat`  
Change GTVIRT mode (Sleep or Wake\_up), and returns previous mode.
- `gtv_newimage`  
Create a new image slot
- `gtv_image`  
Draw a bitmap to an image slot
- `gtv_numimage`  
Get a free image slot
- `gtv_majimage`  
Define parameters of an image slot
- `gtv_fillpoly`  
Fill a closed polygon.
- `init_gtvirt`  
Initialize the GTVIRT library
- `exit_clear`  
Quick exit, to be used at program completion.
- `run_gtv`  
Dispatch the GTVL commands to appropriate subroutines
- `exec_gtv`  
Execute a GTVL language command or command procedure

*Immediate* routines are used to produce immediate actions. They use the immediate pen. These subroutines are

- `gti_beep`: beep
- `gti_clear`: clear the current window
- `gti_draw`: draw line to current
- `gti_out`: flush the normal drawing buffer.
- `gti_pen`: select the immediate pen
- `gti_poly`: draw a polyline
- `gti_reloc`: relocate the immediate pen
- `gti_where`: returns immediate pen position



# Index

GDF\_CLIS, 44  
GDF\_CRIS, 45  
GDF\_EXIS, 45  
GDF\_FRIS, 44  
GDF\_FRMS, 47  
GDF\_GEIS, 44  
GDF\_GEMS, 46  
GDF\_REIS, 46  
GDF\_STIS, 45  
GDF\_UPMS, 47  
GDF\_WRIS, 46

## ROUTINES

EXEC\_COMMAND, 9  
EXEC\_PROGRAM, 9  
FREE\_VM, 19  
GAGOUT, 9  
LENC, 19  
SIC\_AMBIGS, 16  
SIC\_ANALYSE, 17  
SIC\_BEGIN, 4, 10, 14  
SIC\_CH, 8, 16  
SIC\_CTRL, 15  
SIC\_DEF\_CHAR, 10  
SIC\_DEF\_CHARS, 10  
SIC\_DEF\_DBL, 10  
SIC\_DEF\_FUNC, 13  
SIC\_DEF\_INTE, 10  
SIC\_DEF\_LOGI, 10  
SIC\_DEF\_LOGIN, 10  
SIC\_DEF\_REAL, 10  
SIC\_DEF\_STRN, 10  
SIC\_DEFSTRUCTURE, 10  
SIC\_DELSYMBOL, 19  
SIC\_DELVARIABLE, 13  
SIC\_DESCRIPTOR, 11  
SIC\_FORMAT, 18  
SIC\_GET\_CHAR, 11  
SIC\_GET\_DBL, 11  
SIC\_GET\_INTE, 11

SIC\_GET\_LOGI, 11  
SIC\_GET\_REAL, 11  
SIC\_GETLUN, 9  
SIC\_GETSYMBOL, 19  
SIC\_GETVM, 19  
SIC\_I4, 8, 16  
SIC\_INCARNATE, 12  
SIC\_INSERT, 14  
SIC\_INTER.STATE, 15  
SIC\_KE, 8, 16  
SIC\_L4, 8, 16  
SIC\_LANG, 15  
SIC\_LEN, 17  
SIC\_LET\_CHAR, 11  
SIC\_LET\_DBL, 11  
SIC\_LET\_INTE, 11  
SIC\_LET\_LOGI, 11  
SIC\_LET\_REAL, 11  
SIC\_LIRE, 14, 15  
SIC\_LOG, 15  
SIC\_LOGICAL, 13  
SIC\_LOWER, 18  
SIC\_MATERIALIZE, 12  
SIC\_MATH, 13  
SIC\_NARG, 16  
SIC\_NEXT, 17  
SIC\_OPT, 19  
SIC\_PARSE, 18  
SIC\_PRESENT, 16  
SIC\_R4, 8, 16  
SIC\_R8, 8, 16  
SIC\_RUN, 19  
SIC\_SETSYMBOL, 19  
SIC\_SEXA, 13  
SIC\_START, 17  
SIC\_UPPER, 18  
SIC\_VOLATILE, 12  
SIC\_WPR, 18  
SIC\_WPRN, 18  
VAR\_TYPE, 11